HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Cardiac/Pulmonary



NOTICE

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁑ . When you see ⁑ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■ , followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01♦LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	***01 LBL^T SAMPLE***
02 "THIS IS A "	[ALPHA] ***THIS IS A*** [ALPHA]	***02^T THIS IS A***
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	***03^T ⊢ SAMPLE***
04 AVIEW	■ AVIEW [ALPHA]	***04 AVIEW***
05 6	6	***05 6***
06 ENTER↑	[ENTER↑]	***06 ENTER ↑***
07 -2	2 [CHS]	***07 -2***
08 /	[÷]	***08 /***
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	***09 ABS***
10 STO IND L	[STO] ■ [•] L	***10 STO IND L***
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	***11^TR3=***
12 ARCL 03	■ [AVIEW]	***12 ARCL 03***
13 AVIEW	[ALPHA]	***13 AVIEW***
14 RTN	■ [RTN]	***14 RTN***

TABLE OF CONTENTS

PULMONARY FUNCTIONS/VITAL CAPACITY

This program provides calculation of predicted and percent predicted values of the following functions:

VC = Vital capacity in liters.

FEV_1 = Forced expiratory volume after one second in liters.

MEFR = Maximum expiratory flow rate in liters/second.

MVV = Maximum ventilatory volume after 12 seconds in liters.

RV = Residual volume in liters.

TLC = Total lung capacity in liters.

FRC = Functional residual capacity in liters.

FEF = Forced expiratory flow from 25% to 75% ($FEF_{25\%-75\%}$) in liters/sec.

MALE	FEMALE
VC = (.058•Ht)-(.025•age)-4.24	VC = (.045•Ht)-(.024•age)-2.852
FEV_1 = (.036•Ht)-(.032•age)-1.26	FEV_1 = (.035•Ht)-(.025•age)-1.932
MEFR = (.043•Ht)-(.047•age)+2.07	MEFR = (.057•Ht)-(.036•age)-2.532
MVV = (.9•Ht)-(1.51•age)+27	MVV = (.762•Ht)-(.81•age)-6.29
RV = (.03•Ht)+(.015•age)-3.75	RV = (.024•Ht)+(.012•age)-2.63
TLC = (.094•Ht)-(.015•age)-9.17	TCL* = (.078•Ht)-(.01•age)-7.36
FRC = (.051•Ht)-5.05	FRC = (.047•Ht)-4.86
FEF = (.02•Ht)-(.04•age)+2	FEF = (.02•Ht)-(.03•age)-(.00006•age^2)+1.3

where Ht is in cm and age in years.

$$\text{Actual FEF} = (.5\cdot VC)/\Delta t$$

$$\text{where } \Delta t = t_{75\%VC} - t_{25\%VC}$$

$$25\%VC = .25VC$$

$$75\%VC = .75VC$$

NOTE: This program requires one extra memory module in the HP-41C because of its length.

*For females, if height is greater than 174 cm, 1 cm is added to the height before TLC is calculated.

References: This program is based on HP-67/97 Users' Library programs and on the HP-65 Medical Pac I.

Morris, J.F., Koski, A., & L.C. Johnson, *AM. REV. RESP. DIS.*, 57: 103 (1971).

Bates et.al., *RESP. FTN, IN DISEASE*, Saunders (1971).

Example:

For a male patient, height 72 in., age 28 the measured VC = 5.2 ℓ. Calculate all predicted levels and % predicted for VC and FEF ($t_{25\%} = .4$, $t_{75\%} = 1.0$).

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] VITCAP [ALPHA]	M/F?
M [R/S]	HT=?
72 [CHS] [R/S]	AGE=?
28 [R/S]	28.00
5.2 [A]	VC=5.67
[R/S]	%PRED=91.76
[B]	FEV1=4.43
[C]	MEFR=8.62
[D]	MVV=149.31
[E]	RV=2.16
[F]	TLC=7.60
[G]	FRC=4.28
[H]	FEF=4.54
[R/S]	25%VC=1.30
[R/S]	T25%=?
.4 [R/S]	75%VC=3.90
[R/S]	T75%=?
1 [R/S]	ACT FEF=4.33
[R/S]	%PRED=95.50

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program.			
2	Begin program execution.		[XEQ] VITCAP	M/F?
3	Input M for make, F for female.	M or F	[R/S]	HT=?
4	Input patient height (cm or — inches).	cm or -in	[R/S]	AGE=?
5	Input patient age.	yrs	[R/S]	
6	Input measured value (optional) and			
	calculate:			
	Predicted VC	(VC_{meas})	[XEQ] VC	
			or [A]	VC=
	Predicted FEV_1	(FEV_{meas})	[XEQ] FEV1	
			or [B]	FEV1=
	Predicted MEFR	($MEFR_{meas}$)	[XEQ] MEFR	
			or [C]	MEFR=
	Predicted MVV	(MVV_{meas})	[XEQ] MVV	
			or [D]	MVV=
	Predicted RV	(RV_{meas})	[XEQ] RV	
			or [E]	RV=
	Predicted TLC	(TLC_{meas})	[XEQ] TLC	
			or [F]	TLV=
	Predicted FRC	(FRC_{meas})	[XEQ] FRC	
			or [G]	FRC=
7	(Optional) If measured value was input			
	obtain % predicted value.		[R/S]*	%PRED=
8	Input measured VC (not necessary if input	(VC_{meas})	[XEQ] FEF	
	above) and calculate predicted FEF.		or [H]	FEF=
	*This [R/S] not needed if calculator is used with printer.			

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
8a	and 25% VC		[R/S]*	25%VC=
8b	Input time associated with 25% VC from		[R/S]*	T25%=?
	spirogram and display 75% VC.	T@25%VC	[R/S]	75%VC=
8c	Input time associated with 75% VC and		[R/S]*	T75%=?
	display actual FEF followed by % predicted.	T@75%VC	[R/S]	ACT FEF=
			[R/S]*	%PRED=
9	For a new calculation go to step 2.			
	*This [R/S] not needed if calculator is used with printer.			

Program Listings

Program	Comments
01♦LBL "VIT CAP"	
02 SF 00	
03 CF 01	
04 CF 02	Initialize
05 SF 21	
06 SF 27	
07 FIX 2	- - - - - - - - - -
08 "M"	
09 ASTO Y	
10 "M/F?"	Male or female?
11 AON	
12 PROMPT	
13 AOFF	
14 ASTO X	M or F Male
15 CLA	- - - - - - - - - -
16 X=Y?	
17 CF 00	
18 "HT=?"	Input HT
19 PROMPT	CM or inches
20 X>0?	
21 GTO 00	
22 2.54	Convert to CM
23 *	
24 CHS	- - - - - - - - - -
25♦LBL 00	
26 STO 00	
27 "AGE=?"	Input age
28 PROMPT	- - - - - - - - - -
29 STO 01	
30 STOP	
31♦LBL A	Calculate vital
32♦LBL "VC"	capacity
33 "VC"	VC Input
34 FS? 22	
35 STO 02	
36 FS? 00	Female?
37 GTO 00	Yes
38 .058	Male constants
39 STO 05	
40 .025	
41 STO 06	
42 4.24	
43 STO 07	
44 GTO 01	
45♦LBL 00	Female constants
46 .045	
47 STO 05	
48 .024	
49 STO 06	
50 2.852	
51 STO 07	Go to calculation
52 GTO 01	routine
53♦LBL B	- - - - - - - - - -
54♦LBL "FEV1"	Calculate FEV1
55 "FEV1"	
56 FS? 00	
57 GTO 00	
58 .036	
59 STO 05	
60 .032	
61 STO 06	
62 1.26	
63 STO 07	
64 GTO 01	
65♦LBL 00	
66 .035	
67 STO 05	
68 .025	
69 STO 06	
70 1.932	
71 STO 07	
72 GTO 01	- - - - - - - - - -
73♦LBL C	Calculate MEFR
74♦LBL "MEFR"	
75 "MEFR"	
76 FS? 00	
77 GTO 00	
78 .043	
79 STO 05	
80 .047	
81 STO 06	
82 -2.07	
83 STO 07	
84 GTO 01	
85♦LBL 00	
86 .057	
87 STO 05	
88 .036	
89 STO 06	
90 2.532	
91 STO 07	- - - - - - - - - -
92♦LBL 01	Calculation
93 CF 03	subroutine
94 R↑	
95 RCL 00	
96 FS?C 01	
97 XEQ 08	

Program Listings

98 RCL 05	
99 *	
100 RCL 01	
101 RCL 06	
102 *	
103 -	
104 RCL 07	
105 -	
106 FS? 02	FEF Calc.?
107 XEQ 09	Yes
108 FC?C 22	Was measured value input?
109 SF 03	
110 XEQ 10	
111 FS?C 02	FEF Calc.?
112 RTN	Yes, stop
113 FS? 03	No measured value
114 STOP	input. Stop
115◆LBL 07	Calc. % of
116 /	predicted value
117 1 E2	
118 *	
119 "% PRED"	
120◆LBL 10	Output subroutine
121 "⊢="	
122 ARCL X	
123 AVIEW	
124 RTN	
125◆LBL D	Calculate MVV
126◆LBL "MVV "	
127 "MVV"	
128 FS? 00	
129 GTO 00	
130 .9	
131 STO 05	
132 1.51	
133 STO 06	
134 27	
135 CHS	
136 STO 07	
137 GTO 01	
138◆LBL 00	
139 .762	
140 STO 05	
141 .81	
142 STO 06	
143 6.29	
144 STO 07	
145 GTO 01	
146◆LBL E	Calculate RV
147◆LBL "RV"	
148 "RV"	
149 FS? 00	
150 GTO 00	
151 .03	
152 STO 05	
153 -.015	
154 STO 06	
155 3.75	
156 STO 07	
157 GTO 01	
158◆LBL 00	
159 .024	
160 STO 05	
161 -.012	
162 STO 06	
163 2.63	
164 STO 07	
165 GTO 01	
166◆LBL F	
167◆LBL "TLC "	Calculate TLC
168 "TLC"	
169 FS? 00	
170 GTO 00	
171 .094	
172 STO 05	
173 .015	
174 STO 06	
175 9.17	
176 STO 07	
177 GTO 01	
178◆LBL 00	
179 174	
180 RCL 00	
181 X>Y?	
182 SF 01	
183 .078	
184 STO 05	
185 .01	
186 STO 06	
187 7.36	
188 STO 07	
189 GTO 01	
190◆LBL G	Calculate FRC
191◆LBL "FRC "	
192 "FRC"	
193 CLX	
194 STO 06	
195 FS? 00	

Program Listings

```
196 GTO 00
197 .051
198 STO 05
199 5.05
200 STO 07
201 GTO 01
202♦LBL 00
203 .047
204 STO 05
205 4.86
206 STO 07
207 GTO 01
                          - - - - - - - - - - -
208♦LBL H                 Calculate FEF
209♦LBL "FEF
"
210 "FEF"
211 SF 02
212 .02
213 STO 05
214 FS? 00
215 GTO 00
216 .04
217 STO 06
218 -2
219 STO 07
220 GTO 02
221♦LBL 00
222 .03
223 STO 06
224 -1.3
225 STO 07
226♦LBL 02
227 XEQ 01
228 STO 04
229 RCL 02
230 4
231 /
232 "25% VC"                Output 25% VC
233 XEQ 10
234 "T25%=?"                Input time at
235 PROMPT                  25% VC
236 STO 03
237 X<>Y
238 3
239 *
240 "75% VC"                Output 75% VC
241 XEQ 10
242 "T75%=?"                Input time at
243 PROMPT                  75% VC
244 RCL 03
245 -
```

```
246 RCL 02
247 2
248 /
249 X<>Y
250 /
251 "ACT FEF                Output actual FEF
"
252 XEQ 10
253 RCL 04                  Calc. % predicted
254 GTO 07                  FEF Calc.
255♦LBL 09                  Male?
256 FC? 00
257 RTN                     Yes, RTN.
258 6 E-5                   Female, alternate
259 RCL 01                  calculation.
260 X↑2
261 *
262 -
263 RTN
264♦LBL 08                  TLC, female
265 1                       >174 cm.
266 +
267 END
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Height (cm)	50	
	Age (yrs)		
	VC_{meas}		
	$t_{25\%}$		
	FEF_{PRED}		
05	Coeff. A	55	
	Coeff. B		
	Coeff. C		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 96 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	S	Female	Male
01	C	Female 174 cm	Other
02	C	FEF Calc.	Other
03	C	Meas. value input	No meas. value
21	S	Display & Printer control	
22	C	Data entry	No data entry
27	S	User mode	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

LUNG DIFFUSION AND ARTERIAL CO_2 NORMALIZATION

Lung Diffusion: This portion of the program evaluates the equation to calculate the lung diffusion capacity (DLCO) using the single breath method.

Equation used:

$$DLCO = \frac{V_A(0.084)}{BHT} \ln \frac{F_A CAR}{F_I CAR} \frac{0.3}{F_A CO}$$

Note: The initial concentration of carbon monoxide ($F_I CO$) is assumed to be 0.3%. If a different standard value for $F_I CO$ is desired, it may be entered.

Reference: Comroe, et.al., *The Lung*, Year Book Medical Publishers Inc., 1962.

Arterial CO_2 Normalization: This portion of the program calculates the additional dead space (DS_{add}) needed in a hypocapnic ventilator patient's breathing circuit to raise the arterial CO_2 partial pressure ($P_a CO_2$) to 40 millimeters of mercury (mmHg).

Equations used:

$$DS_{add} = \frac{TV - DS}{40 - \Delta P_{CO_2}} (40 - P_a CO_2)$$

$$\Delta P_{CO_2} = P_a CO_2 - P_{\bar{E}} CO_2 \quad (\text{or} \quad P_a CO_2 - 5 \quad \text{if } P_{\bar{E}} CO_2 \text{ is not entered})$$

$$TV - DS = TV - [1.47Wt(kg) + DS_p]$$

where DS_p = dead space.

Detailed Instructions for Arterial CO_2 Normalization calculation:

Input the patient's weight in kilograms, or in pounds followed by [CHS]. Then input the $P_a CO_2$ in mmHg. If the patient's lung status is abnormal answer the question LUNG NORMAL? by inputting N and then inputting $P_{\bar{E}} CO_2$ (the mixed expired CO_2 partial pressure). If lung condition is normal answer Y ($P_{\bar{E}} CO_2$ is not required). Then input the present tidal volume and ventilator dead space. The additional rebreathing dead space is calculated. This must be added to the patient's circuit to achieve $P_a CO_2$ normalization.

Warning: The additional dead space required by this program must be inserted into the patient's breathing circuit *without* changing the ventilator rate or tidal volume.

Measure and input the mixed expired CO_2 partial pressure if lung function is abnormal.

References: Suwa, Kunio; Geffin, Bennie; Pontoppidan, Henning; Bendixen, Henry; *"A Nomogram for Dead Space Requirement During Prolonged Artificial Ventilation"*, Anesthesiology, v. 29, 1968 Nov.-Dec.

Lung Diffusion

Example 1:

Calculate the lung diffusing capacity using an initial helium carrier gas concentration of 10%, an alveolar helium concentration of 8%, an alveolar carbon monoxide concentration of 0.159%, an initial carbon monoxide concentration of 0.3%, a breath holding time of 10 seconds, and an alveolar volume of 4930 milliliters.

Example 2:

For the same data, calculate lung diffusing capacity assuming an initial carbon monoxide concentration of 0.45%.

Keystrokes: Example 1	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] DLCO [ALPHA]	FICO=.3?
[R/S]	FICAR=?
10 [R/S]	FACAR=?
8 [R/S]	FACO=?
.159 [R/S]	BHT=?
10 [R/S]	VA=?
4930 [R/S]	DLCO=17.05

Keystrokes: Example 2	Display:
[A]	FICO=.3?
.45 [R/S]	FICAR=?
10 [R/S]	FACAR=?
8 [R/S]	FACO=?
.159 [R/S]	BHT=?
10 [R/S]	VA=?
4930 [R/S]	DLCO=33.84

Arterial CO_2 Normalization

Example:

Calculate the additional dead space required by a 50 kilogram patient with a P_aCO_2 of 25 mmHg with normal lung status having a tidal volume of 900 ml and a present dead space of 25 ml.

Keystrokes:	Display:
[XEQ] [ALPHA] NORM [ALPHA]	WT=?
50 [R/S]	PaCO2=?
25 [R/S]	LUNG NORMAL?
Y [R/S]	TV=?
900 [R/S]	DSP=?
25 [R/S]	DSadd=343.50

User Instructions

				SIZE: 012
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program. Go to step 2 for lung			
	diffusion (DLCO). Go to step 6 for			
	Arterial CO_2 normalization.			
	LUNG DIFFUSION:			
2	Begin the lung diffusion calculation.		[XEQ] DLCO	FICO=.3?
3	If initial concentration of carbon monoxide			
	is other than 0.3% input correct value.	$F_I CO$	[R/S]	FICAR=?
	Otherwise, press [R/S] *without data entry*.		[R/S]	FICAR=?
4	Input initial concentration of carrier and	$F_I CAR$	[R/S]	FACAR=?
	alveolar carrier concentration.	FACAR	[R/S]	FACO=?
5	Input the arterial concentration of carbon			
	monoxide, the breath holding time (in	FACO	[R/S]	BHT=?
	seconds) and the alveolar volume. The lung	BHT	[R/S]	VA=?
	diffusion capacity (DLCO) is calculated.	VA	[R/S]	DLCO=?
	ARTERIAL CO_2 NORMALIZATION:			
6	Begin the normalization calculation.		[XEQ] NORM	WT=?
7	Input the patient weight			
	-in Kg or,	Kg	[R/S]	
	-in pounds.	lb.	[CHS] [R/S]	PaCO2=?
8	Input the arterial CO_2 partial pressure.	$PaCO_2$	[R/S]	LUNG NORMAL?
9a	If lung is normal answer "Y" and go to			
	step 10.	Y	[R/S]	TV=?
9b	If lung is abnormal answer "N" and then	N	[R/S]	PeCO2=?
	input $PeCO_2$ (mixed expired CO_2 partial	$PeCO_2$	[R/S]	TV=?
	pressure).			
10	Input the tidal volume and present dead	TV(ml)	[R/S]	DSP=?
	space. The additional dead space	DSp(ml)	[R/S]	DSadd=

User Instructions

SIZE:

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	requirement is calculated.			
11	For a new DLCO calculation press [A] and			
	go to step 3.		[A]	FICO3=.3?
12	For a new CO_2 normalization press [B] and			
	go to step 7.		[B]	WT=?

Program Listings

Program step	Comment
01♦LBL "DLCO"	DLCO Calc
02♦LBL A	Initialization
03 FIX 2	
04 SF 21	
05 SF 27	
06 .3	
07 "FICO=.3?"	Is F_ICO=.3%?
08 PROMPT	Input other F_ICO
09 STO 00	
10 "FICAR=?"	Input F_I CAR
11 PROMPT	&
12 STO 01	
13 "FACAR=?"	F_A CAR
14 PROMPT	&
15 STO 02	
16 "FACO=?"	F_A CO
17 PROMPT	
18 STO 03	&
19 "BHT=?"	Breath hold-time
20 PROMPT	
21 STO 08	&
22 "VA=?"	Alveolar volume
23 PROMPT	
24 .084	
25 *	
26 X<>Y	Calculate DLCO
27 /	
28 RCL 02	
29 RCL 01	
30 /	
31 RCL 00	
32 RCL 03	
33 /	
34 *	
35 LN	
36 *	
37 "DLCO"	Display DLCO
38 GTO 10	CO_2 normalization
39♦LBL "NORM"	
40♦LBL B	
41 FIX 2	Initialization
42 SF 21	
43 SF 27	
44 "WT=?"	Input weight

Program step	Comment
45 PROMPT	X + or -
46 X>0?	+
47 GTO 00	-
48 CHS	Convert to kg
49 2.205	
50 /	
51♦LBL 00	
52 STO 06	Sto wt in kg
53 "PaCO2=?"	Input $PaCO_2$
54 PROMPT	
55 STO 01	Calc. normal
56 5	$PeCO_2$ & store
57 -	
58 STO 08	
59 "Y"	
60 ASTO Y	
61 "LUNG NORMAL?"	LUNG NORMAL?
62 AON	
63 PROMPT	
64 AOFF	
65 ASTO X	Yes or no
66 X=Y?	Yes, normal
67 GTO 01	No, input $PeCO_2$
68 "PeCO2=?"	
69 PROMPT	
70 STO 08	
71♦LBL 01	Input tidal vol.
72 "TV=?"	
73 PROMPT	&
74 STO 07	Present dead space
75 "DSP=?"	
76 PROMPT	
77 RCL 06	
78 1.47	Calculate
79 *	
80 +	additional
81 RCL 07	
82 X<>Y	dead space
83 -	
84 40	
85 RCL 01	
86 RCL 08	
87 -	
88 -	
89 /	
90 40	
91 RCL 01	

Program Listings

92 -	
93 *	
94 "DSadd"	Display DS add
95♦LBL 10	
96 "⊢="	Display routine
97 ARCL X	
98 AVIEW	
99 END	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	.3 or F CO	50	
	F CAR/$PaCO_2$		
	F CAR		
	F CO		
05		55	
	WT(Kg)		
	TV		
	BHT/$PeCO_2$		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 44 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21	S	Print & Display	Pause only
27	S	User Mode	Reg. Mode

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

VENTILATOR SETUP AND CORRECTIONS

This program calculates the initial tidal volume for a ventilator patient. The first part calculates an approximation to the Radford nomogram tidal volume with correction for ventilator dead space only. The second part corrects the tidal volume for altitude, patient's temperature, daily activity, use of a tracheotomy tube, and metabolic acidosis in anesthesia.

Equations Used:

V_A = Alveolar minute volume = $10^{(C_1 \text{LOG WT} + C_2)/100}$ ml/min

TV_A = Alveolar tidal volume = $\frac{V_A}{r}$ ml

TV_{bas} = Basal tidal volume = $(V_{T_A} + \text{Wt (lbs)})$ ml

TV_{corr} = Basal tidal volume + ventilator dead space

where:

r = Breathing rate (breaths per minute)

For Females:

$C_1 = \begin{cases} 124; & Wt \leqslant 8kg \\ 61; & 8kg < Wt \leqslant 23kg \\ 44.2; & Wt > 23kg \end{cases}$

$C_2 = \begin{cases} 193; & Wt \leqslant 8kg \\ 249; & 8kg < Wt \leqslant 23kg \\ 272; & Wt > 23kg \end{cases}$

For Males:

$C_1 = \begin{cases} 124; & Wt \leqslant 8kg \\ 61; & Wt > 8kg \end{cases}$

$C_2 = \begin{cases} 193; & Wt \leqslant 8kg \\ 249; & Wt > 8kg \end{cases}$

Corrections:

Temperature: +5% per °F above 99° (rectal)

Altitude: +5% per 2000' above sea level

Activity: +10%

Tracheotomy: -½ body weight in pounds

Metabolic acidosis in anesthesia: +20%

Reference: Radford, Edward P., "*Ventilation Standards for Use in Artificial Respiration*", Journal of Applied Physiology, 7:451, 1955.

Warning:

•This program yields an approximation to the Radford nomogram. The nomogram may not be applied with confidence to patients with muscular activity or abnormal lung function.

•Apply only the corrections which pertain to the patient for whom the program is being run.

Example:

1) Calculate the predicted tidal volume for a 170 pound comatose male having a breath rate of 15 breaths per minute, ventilator dead space of 25 milliliters, fever of 101°F, who is located 500 feet above sea level.

2) What would be the corrected tidal volume if this patient were in metabolic acidosis?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] VENT [ALPHA]	WT=?
170 [CHS] [R/S]	M/F?
M [R/S]	BR=?
15 [R/S]	BASAL TV=461.74
[R/S]	DSV=?
25 [R/S]	DSV CORR TV=486.74
[R/S]	BT=?
101 [CHS] [R/S]	TEMP CORR TV=535.42
[R/S]	ALT=?
500 [CHS] [R/S]	ALT CORR TV=542.11
[XEQ] [ALPHA] METACID [ALPHA]	ACIDOSIS CORR TV=650.53

User Instructions

				SIZE: 009
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program and begin Ventilator			
	Setup and Correction calculations.		[XEQ] VENT	WT = ?
2	Input the patient weight in Kg or,	Kg	[R/S]	
	in pounds.	lb	[CHS] [R/S]	M/F ?
3	If patient is a male input M or, if a			
	female, F.	M or F	[R/S]	BR=?
4	Input the breathing rate and calculate			
	basal tidal volume.	BR/min	[R/S]	BASAL TV=
5	To correct for ventilator dead space:			
	Input ventilator dead space and calculate		[R/S]*	DSV=?
	dead space corrected tidal volume.	$DS_{v,ml}$	[R/S]	DSV CORR TV=
6	To correct for temperature:		[R/S]*	BT=?
	Input patient temperature in °C	°C	[R/S]	
	(or °F [CHS]).	°F	[CHS] [R/S]	TEMP CORR TV=
7	To correct for altitude: Input the		[R/S]*	ALT=?
	altitude in meters	m	[R/S]	
	(or, in feet).	ft	[CHS] [R/S]	ALT CORR TV=
	The following steps are optional:			
8	To correct for minor daytime activity		[XEQ] ACTIVE	
	(noncomatose patient).		or [B]	ACTIVITY CORR TV=
9	To correct for use of a tracheotomy tube		[XEQ] TRACH	
			or [C]	TRACH CORR TV=
10	To correct for metabolic acidosis during		[XEQ] METACID	
	anesthesia		or [D]	ACIDOSIS CORR TV=
11	For a new case, press [A] and go to step 2.		[A]	WT = ?
	*This [R/S] not needed if calculator is used with printer.			

Program Listings

01♦LBL "VENT"	Ventilator corr.
02 FIX 2	Initialize
03 SF 21	
04 SF 27	
05♦LBL A	Input wt.
06 "WT=?"	
07 PROMPT	
08 X>0?	
09 GTO 00	
10 -2.205	Convert to kg
11 /	
12♦LBL 00	
13 STO 06	STO wt
14 "M"	Male or female?
15 ASTO Y	
16 "M/F?"	
17 CF 01	
18 AON	
19 PROMPT	
20 AOFF	
21 ASTO X	X="M"?
22 X=Y?	
23 SF 01	Yes, male
24 "BR=?"	Input breathing rate
25 PROMPT	
26 STO 08	Female?
27 FC?C 01	Yes, GTO female calc.
28 GTO 01	
29♦LBL 05	Male calc.
30 8	
31 RCL 06	
32 X<=Y?	
33 GTO 03	
34 249	Input constants
35 ENTER↑	
36 61	
37 GTO 04	
38♦LBL 03	
39 193	
40 ENTER↑	
41 124	
42 GTO 04	
43♦LBL 01	Female constants
44 23	
45 RCL 06	
46 X<=Y?	
47 GTO 05	
48 272	
49 ENTER↑	
50 44.2	
51♦LBL 04	Common male/female calc.
52 RCL 06	
53 LOG	
54 *	
55 +	
56 E2	
57 /	
58 10↑X	
59 RCL 08	
60 /	
61 RCL 06	
62 2.205	
63 *	
64 +	Output basal TV
65 STO 01	
66 "BASAL"	
67 XEQ 10	Input dead space
68 "DSV=?"	
69 PROMPT	
70 ST+ 01	
71 RCL 01	Output DSV corr.
72 "DSV"	
73 XEQ 07	Input body temp.
74 "BT=?"	
75 PROMPT	
76 X<0?	
77 GTO 01	Convert to °F
78 1.8	
79 *	
80 32	
81 +	
82 CHS	
83♦LBL 01	
84 CHS	BT-99°
85 99	
86 -	
87 X<=0?	No corr. if BT≤99°F
88 GTO 02	
89 .05	Correct for BT>99°F
90 *	
91 RCL 01	
92 *	
93 ST+ 01	
94♦LBL 02	
95 RCL 01	
96 "TEMP"	Output temp. TV
97 XEQ 07	
98 "ALT=?"	Input altitude
99 PROMPT	

Program Listings

```
100 X<0?
101 GTO 02              Convert to feet
102 -3.28
103 *
104♦LBL 02              Altitude corr.
105 CHS
106 2 E3
107 /
108 RCL 01
109 *
110 .05
111 *
112 ST+ 01
113 RCL 01
114 "ALT"
115 GTO 07              Output ALT corr.
                        TV
116♦LBL "ACTIVE"        Activity correc.
117♦LBL B
118 RCL 01
119 .1
120 *
121 ST+ 01
122 RCL 01
123 "ACTIVITY"
124 GTO 07
125♦LBL "TRACH"         Tracheotomy corr.
126♦LBL C
127 RCL 01
128 RCL 06
129 1.1023
130 *
131 -
132 STO 01
133 "TRACH"
134 GTO 07
135♦LBL "METACID"       Metabolic acido-
                        sis corr.
136♦LBL D
137 RCL 01
138 .2
139 *
140 ST+ 01
141 RCL 01
142 "ACIDOSIS"
143♦LBL 07
144 "⊢ CORR"            Output corr. TV
145♦LBL 10              Output routing
146 "⊢ TV="
147 ARCL X
148 AVIEW
149 END
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	TV		
05		55	
	WT(kg)		
	BR		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 58 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01	C	Male	Female
21	S	Print & Display	Pause only
27	S	User mode	Reg. mode

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BLOOD CHEMISTRY I

BLOOD ACID - BASE STATUS
VIRTUAL PO_2 AND O_2 SATURATION AND CONTENT

These two programs perform various related blood chemistry and blood gas calculations.

Blood-Acid Base Status

This program computes total plasma CO_2 and base excess from PCO_2, pH and hemoglobin concentration.

Equations:

Total plasma CO_2 is calculated from the Henderson-Hasselbalch equation:

$$TCO_2 = s \cdot PCO_2[1 + 10^{pH-pK}]$$

where

TCO_2 = total CO_2 in plasma, mmol/ℓ

s = solubility of CO_2 in plasma, mmol/ℓ (taken to be 0.0307)

PCO_2 = partial pressure of CO_2 in the blood, mmHg

pK = 6.11

This does not take into account the small temperature dependence of both s and pK, nor the pH dependence of pK. For this reason the formula for TCO_2 will be most accurate if 37° C values for for PCO_2 and pH are used.

The base excess is calculated from an equation suggested by Siggaard-Andersen:

$$[BE]_b = (1 - 0.0143Hgb) \cdot ([HCO_3] - (9.5 + 1.63\,Hgb)(7.4 - pH) - 24)$$

where

$[BE]_b$ = Base Excess in meq/ℓ of blood

Hgb = Hemoglobin concentration in g/100 ml

and plasma $[HCO_3]$ is calculated from the Henderson-Hasselbalch equation in the form:

$$[HCO_3] = s \cdot PCO_2 \cdot 10^{pH-pK}$$

Siggaard-Andersen used 38°C values for PCO_2 and pH. Only small errors will result from using 37°C values, but body temperature corrected values should not be used if the patient has any significant hyper or hypothermia. In only body temperature values are know, the "Anaerobic PCO_2 and pH change" program may be used to correct them back to 37°C. (See special instructions for that program).

NOTE: While Thomas has shown that this equation may produce large errors for very abnormal conditions, it matches the Siggaard-Andersen nomogram for $[BE]_b$, to within ± 1 meq/ℓ in most cases.

VIRTUAL PO_2 AND O_2 SATURATION CONTENT:

The first part of this program computes virtual PO_2 for use in estimating O_2 saturation. Generally, it will be more convenient to calculate venous values first, as arterial values are frequently needed in other programs and, thus, will be left in the storage registers after both calculations.

The equation solved is:

$$VPO_2 = PO_2 \cdot 10^{[0.024(37-BT) + 0.48)pH-7.4) + 0.06(\log PCO_2)]}$$

which is a hybrid of the equation used by Thomas and that used by Kelman. There is some disagreement regarding the best value of the pH multiplier, 0.48 being used by most workers, but see, for example, Kelman.

The second part of the program estimates O_2 saturation of blood from virtual PO_2 and computes O_2 content. If the actual O_2 saturation is known, O_2 content may be computed directly.

EQUATIONS:

The part of the program for estimating O_2 saturation is based on the polynomial curve fit of Thomas, where VPO_2 is in mmHg.

$$O_2\ Sat = \frac{(VPO_2)^4 - 15(VPO_2)^3 + 2045(VPO_2)^2 + 2000(VPO_2)}{(VPO_2)^4 - 15(VPO_2)^3 + 2400(VPO_2)^2 + 31{,}100(VPO_2) + 2{,}400{,}000}$$

This calculation assumes that the oxygen dissociation curve for the hemoglobin is normal. The O_2 content is computed from:

$$C_xO_2(\text{Vol.\%}) = 1.34 \cdot \frac{SAT(\%)}{100} \cdot Hgb(g/100ml) + 0.0031\, PO_2\ (mmHg)$$

NOTE: Virtual PO_2 is not in any way a real physiologic PO_2. Its only function is for use in estimating O_2 saturation, and it should never be confused with PO_2 corrected to body temperature. Furthermore, it must always be calculated from blood parameters measured at or corrected to 37°C. The calculation will give inaccurate results for fetal hemoglobin, present in babies less than six months old, and for some abnormal adult hemoglobins and certain other blood conditions. The results of the estimation and any subsequent calculations based on it should be viewed with caution unless the dissociation curve has been previously established to be normal. If both PO_2 and O_2 saturation are measured, the program may be used as a convenient means to check for the normality of the dissociation curve.

References:

Siggaard-Andersen, "*Titratable Acid or Base of Body Fluids*", Annals New York Academy of Sciences, 133: 41-48, 1966.

Thomas, L.J. Jr., "*Algorithm for Selected Blood Acid-Base and Blood Gas Calculations*", J. Appl. Physiol., 33: 154-158, 1972.

Kelman, G. Richard, "*Digital Computer Subroutine for the Conversion of Oxygen Tension into Saturation*", J. Appl. Physiol., 21: 1375-1376, 1966.

Example 1:

From the following patient data calculate total plasma CO_2, base excess, and plasma $[HCO3]^-$. Also calculate virtual PO_2 and estimated O_2 saturation and content. Store the value as venous O_2 content.

PO_2 = 75 mmHg	BT - 40°C
PCO_2 = 45 mmHg	Hgb = 16 g/100ml
pH = 7.35	

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] ACID [ALPHA]	PCO2=0.00?
45 [R/S]	PH=0.00?
7.35 [R/S]	HGB=0.00?
16 [R/S]	TCO2=25.39
[R/S]	BE=-1.36
[R/S]	HCO3-=-24.01
[XEQ] [ALPHA] PO2 [ALPHA]	PO2=0.00?
75 [R/S]	PCO2=45.00?
[R/S]	PH=7.35?
[R/S]	BT=0.00?
40 [R/S]	VPO2=59.70
[R/S]	%SAT=90.92
[D]	%SAT=90.92
[R/S]	HGB=16.00?
[R/S]	02 CONT=19.68
[///] [E]	19.68 (stored as venous)

Example 2:

Assuming that VPO2 is actually 75 mmHg, calculate the estimated O_2 saturation and O_2 content.

Keystrokes:	Display:
75 [C]	% SAT=95.08
[D]	% SAT=95.08
[R/S]	HGB=16.00?
[R/S]	02 CONT=20.62

User Instructions

SIZE: 012

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program. For Blood-Acid Base			
	Status calculation, go to step 2. For			
	Virtual PO_2 and O_2 Saturation and Content			
	calculation, go to step 4.			
	BLOOD ACID-BASE STATUS:			
2	Begin the program.		[XEQ] ACID	PCO2=( )?
	Values for the following variables are			
	recalled. If correct, press [R/S] *without*			
	data entry. Otherwise, input the correct			
	value and press [R/S].			
	PCO2	(mmHg)	[R/S]	PH=( )?
	pH	pH	[R/S]	HGB=( )?
	Hemoglobin	Hgb (g/100ml)	[R/S]	TCO2=
	When these entries are completed TCO2 (in			
	mmol/ℓ) is calculated.			
3	Calculate base excess and, if desired,		[R/S]*	BE=(meg/ℓ)
	bicarbonate concentration.		[R/S]	HCO3-=mmol/ℓ
	VIRTUAL PO_2 and O_2 SATURATION AND CONTENT:			
4	Begin the program.		[XEQ] PO2	PO2=( )?
	Values for the following variables are			
	recalled. If correct, press [R/S] *without*			
	prior data entry. Otherwise, input the			
	correct value and press [R/S].			
	PCO2	(mmHg)	[R/S]	PH=( )?
	pH	pH	[R/S]	BT=( )?
	Body Temperature	BT(°C)	[R/S]	VPO2=
	*This [R/S] not necessary if calculator is used with printer.			

User Instructions

SIZE:

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	When these entries are completed VPO2 (in			
	mmHg) is calculated.			
5	Calculate estimated % saturation.		[R/S]*	%SAT=
6	To calculate oxygen content in volume %.		[D]	%SAT=( )?
6a	Input correct % saturation, or press [R/S]	%SAT	[R/S]	HGB=( )?
	if displayed figure is correct. Previously			
	stored hemoglobin is displayed.			
6b	Input correct hemoglobin, or press [R/S]	Hgb (gm/100ml)	[R/S]	02 CONT=(%)
	if displayed hemoglobin is correct. Oxygen			
	content is calculated.			
7	Store oxygen content:			
	As arterial (CaO_2) or,		[E]	(O_2 cont.)
	As venous (CvO_2)		[///]	(O_2 cont.)
8	To calculate a different Blood Acid-base		[A]	PCO2=( )?
	status, press [A] and go to step 2.			
9	To calculate another virtual PO_2 press,		[B]	PO2=( )?
	[B] and go to step 4.			
10	To calculate another % saturation input			
	correct virtual PO_2	VPO_2 (mmHg)	[C]	%SAT=
11.	To calculate another Oxygen content go to			
	step 6.			
	*This [R/S] not necessary if calculator			
	is used with printer.			

Program Listings

Program	Comments
01◆LBL "ACID"	
02 FIX 2	Initialize
03 SF 21	Blood-Acid base
04 SF 27	
05◆LBL A	RCL PCO_2
06 XEQ 02	RCL pH
07 XEQ 03	RCL HGB
08 XEQ 04	
09 RCL 06	Calc. total
10 6.11	plasma CO_2
11 -	
12 10↑X	
13 RCL 05	
14 32.57	
15 /	
16 *	
17 STO 02	
18 LASTX	
19 +	
20 "TCO2"	Display TCO_2
21 XEQ 10	Calc. base excess
22 RCL 09	
23 1.63	
24 *	
25 9.5	
26 +	
27 RCL 06	
28 7.4	
29 -	
30 *	
31 RCL 02	
32 +	
33 24	
34 -	
35 1	
36 RCL 09	
37 70	
38 /	
39 -	
40 *	
41 "BE"	Display B E
42 XEQ 10	Printer?
43 FS? 55	Yes
44 STOP	
45 RCL 02	RCL HCO_3^-
46 "HCO3-"	Display $HCO_3{}^2$
47 GTO 10	Initialize
48◆LBL "PO2"	VPO_2 & O_2 SAT

Program	Comments
49 FIX 2	
50 SF 21	
51 SF 27	
52◆LBL B	RCL PO_2
53 XEQ 01	RCL PCO_2
54 XEQ 02	RCL pH
55 XEQ 03	
56 RCL 11	RCL BT
57 "BT"	
58 XEQ 09	
59 STO 11	Calc. VPO_2
60 37	
61 X<>Y	
62 -	
63 .024	
64 *	
65 RCL 06	
66 7.4	
67 -	
68 .48	
69 *	
70 +	
71 40	
72 RCL 05	
73 /	
74 LOG	
75 .06	
76 *	
77 +	
78 10↑X	
79 RCL 10	
80 *	
81 STO 01	Display VPO_2
82 "VPO2"	
83 XEQ 10	Calc. % SAT
84◆LBL C	
85 STO 01	
86 ENTER↑	
87 ENTER↑	
88 ENTER↑	
89 15	
90 -	
91 *	
92 2045	
93 +	
94 *	
95 2 E3	
96 +	
97 *	
98 STO 08	
99 CLX	
100 15	

Program Listings

101 -	
102 *	
103 2400	
104 ÷	
105 *	
106 31100	
107 -	
108 *	
109 24 E5	
110 +	
111 1 E2	
112 /	
113 ST/ 08	
114 RCL 08	Display % SAT
115 "% SAT"	
116 GTO 10	Calc. O_2 content
117♦LBL D	
118 RCL 08	
119 "% SAT"	RCL % SAT
120 XEQ 09	
121 STO 08	
122 XEQ 04	
123 RCL 08	
124 *	
125 134	
126 *	
127 RCL 01	
128 31	
129 *	
130 +	
131 1 E4	
132 /	
133 "O2 CONT"	Display O_2 cont.
134 GTO 10	
135♦LBL E	Store as C_AO_2
136 STO 04	
137 RTN	
138♦LBL e	Store as C_VO_2
139 STO 03	
140 RTN	
141♦LBL 02	RCL PCO_2
142 RCL 05	
143 "PCO2"	
144 XEQ 09	
145 STO 05	
146 RTN	
147♦LBL 03	RCL pH
148 RCL 06	
149 "PH"	
150 XEQ 09	
151 STO 06	
152 RTN	
153♦LBL 04	RCL HGB
154 RCL 09	
155 "HGB"	
156 XEQ 09	
157 STO 09	
158 RTN	
159♦LBL 01	RCL PO_2
160 RCL 10	
161 "PO2"	
162 XEQ 09	
163 STO 10	
164 RTN	
165♦LBL 10	Display routine
166 "├="	
167 ARCL X	
168 AVIEW	
169 RTN	
170♦LBL 09	RCL routine
171 "├="	
172 ARCL X	
173 "├?"	
174 PROMPT	
175 END	

80	
90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	HCO^-/VPO_2		
	CVO_2		
	CAO_2		
05	PCO_2	55	
	pH		
	Used/%SAT		
	HGB		
10	PO_2	60	
	BT		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 61 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21	S	Print & Display	Pause only
27	S	User mode	Regular

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BLOOD CHEMISTRY II

ANAEROBIC PCO_2 AND pH CHANGE AND ANAEROBIC PO_2 CHANGE

Corrections of PCO_2 and pH for anaerobic temperature change are calculated by this program. In addition, PO_2 measured at 37°C is corrected to body temperature.

Anaerobic PCO_2 and pH Change:

Corrections of PCO_2 and pH for anaerobic temperature change are calculated. The equation for pH is a simplification of a formula from Severinghaus. It ignores the pH and BE dependent terms. This introduces a very small error except at extreme conditions of acid-base status and large temperature shifts. For example, at a pH of 7.2 or 7.6, the error is 0.0013 pH units per °C.

Equations Used:

$$PCO_2(BT) = PCO_2(37) \cdot 10^{0.019(T-37)}$$

$$pH(BT) = pH(37) - 0.0146\ (T-37)$$

Anaerobic PO_2 Change:

This program corrects PO_2, measured at 37°C, to Body Temperature.

Equation Used: Correction of PO_2 for anaerobic temperature change is calculated taking into account the exchange of oxygen between $HgbO_2$ and the dissolved state at high saturation. Below 80% Sat., the relation is approximately

$$\frac{\Delta \text{ Log } PO_2}{\Delta T} = 0.031$$

This factor falls at higher saturations, approaching 0.006 at 100% Sat. The curve given by Severinghaus has been approximated by the following equation in this program:

$$\frac{\Delta \log PO_2}{\Delta T} = \frac{3130-62.5\text{ Sat} + 0.312008\text{ Sat}^2}{100{,}000-1993\text{Sat}+9.9313\text{Sat}^2}$$

Reference: Severinghaus, John W., *Blood Gas Calculator*, J. Appl. Physiol., 21 (3): 1108-1116, 1966.

Detailed User Instructions:

PO_2(BT) replaces the 37°C value in memory with the body temperature value. Therefore, calculation based on the 37°C values in programs for virtual PO_2 and O_2 saturation & content should be accomplished before this program is run. If O_2 saturation has not been measured, it should be estimated by using program for "Virtual PO_2 and O_2 Saturation and Content."

This program may also be used to convert PO_2 between any two temperatures, for example, from body temperature to 37°C. To do this, first determine what the desired temperature change is in °C. Add this to 37°C algebraically, and enter the result as BT. For example, suppose values known at 41°C are to be converted to 37°C. The temperature change is -4°C. Add this to 37°C, resulting 33°C. Executing the program with BT=33°C will then result in the 37°C value for PO_2.

Example:

For a patient with PCO_2 of 45mmHg and a pH of 7.35 at 40°C, calculate corrected values for PCO_2 and pH. If the patient's PO_2 is 75mmHg and % saturation is 90, what is the corrected PO_2?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] ANRB [ALPHA]	0.00
[A]	PCO2=0.00?
45 [R/S]	PCO2=0.00?
7.35 [R/S]	PH=0.00?
40 [R/S]	BT=0.00?
[R/S]	PCO2 CORR.=51.31
[B]	PH CORR.=7.31
90 [R/S]	% SAT=0.00?
75 [R/S]	PO2=0.00?
[R/S]	BT=40.00?
	PO2 CORR.=92.31

User Instructions

SIZE: 012

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program and begin execution.		[XEQ] ANRB	PCO2=( )?
2	For anaerobic PCO_2 and pH change go to			
	step 3. For anaerobic PO_2 go to step 5.			
	ANAEROBIC PCO_2 AND pH CHANGE:			
3	Input correct PCO_2 or press [R/S] if			
	displayed PCO_2 is correct.	PCO_2 (mmHg)	[R/S]	PH=( )?
3a	Input correct pH or press [R/S] if			
	displayed pH is correct.	pH	[R/S]	BT=( )?
3b	Input correct body temperature or press			
	[R/S] if displayed body temperature is			
	correct. Anaerobic PCO_2 is calculated.	BT(°C)	[R/S]	PCO2 CORR.=
4	Calculate pH (BT).		[R/S]	PH CORR.=
	ANAEROBIC PO_2:			
5	Recall % saturation if stored.		[B]	%SAT=( )?
5a	Input correct % saturation or press [R/S]			
	if displayed value is correct.	%SAT	[R/S]	PO2=( )?
5b	Input correct PO_2 or press [R/S] if			
	displayed PO_2 is correct.	PO_2 (mmHg)	[R/S]	BT=( )?
5c	Input correct body temperature or press			
	[R/S] if displayed value is correct.	BT(°C)	[R/S]	PO2 CORR.=
				(mmHg)
6	For a new PCO_2 or pH press [A] and go to			
	step 3. For a new PO_2 go to step 5.			

Program Listings

Program	Comments
01♦LBL "ANRB"	Initialize
02 FIX 2	
03 SF 21	
04 SF 27	
05♦LBL A	
06 RCL 05	RCL PCO_2
07 "PCO2"	
08 XEQ 09	
09 STO 05	
10 RCL 06	RCL pH
11 "PH"	
12 XEQ 09	
13 STO 06	
14 XEQ 04	RCL BT
15 37	
16 -	Calc. PCO_2 corr.
17 STO 01	
18 .019	
19 *	
20 10↑X	
21 RCL 05	
22 *	
23 "PCO2"	Display PCO_2 corr.
24 XEQ 10	
25 RCL 06	Calc. pH corr.
26 RCL 01	
27 .0146	
28 *	
29 -	
30 "PH"	Display pH corr.
31 GTO 10	
32♦LBL B	Calc. PO_2 corr.
33 RCL 08	
34 "% SAT"	RCL % SAT
35 XEQ 09	
36 STO 08	
37 RCL 10	
38 "PO2"	RCL PO_2
39 XEQ 09	
40 STO 10	RCL BT
41 XEQ 04	
42 RCL 08	Calc. PO_2 corr.
43 X↑2	
44 .312008	
45 *	
46 RCL 08	
47 62.5	
48 *	
49 -	
50 3130	
51 +	
52 RCL 08	
53 X↑2	
54 9.9313	
55 *	
56 RCL 08	
57 1993	
58 *	
59 -	
60 1 E5	
61 +	
62 /	
63 RCL 11	
64 37	
65 -	
66 *	
67 10↑X	
68 RCL 10	
69 *	
70 STO 10	
71 "PO2"	Display PO_2 corr.
72♦LBL 10	
73 "⊢ CORR. ="	Display routine
74 ARCL X	
75 AVIEW	
76 RTN	
77♦LBL 04	
78 RCL 11	RCL BT
79 "BT"	
80 XEQ 09	
81 STO 11	
82 RTN	
83♦LBL 09	RCL routine
84 "⊢="	
85 ARCL X	
86 "⊢?"	
87 PROMPT	
88 END	

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	ΔT		
05	PCO_2	55	
	pH		
	%SAT		
10	PO_2	60	
	BT		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 39 USER MODE

ENG ___ FIX 2 SCI ___ ON ___ OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21	S	Print & Display	Pause only
27	S	User mode	Reg. mode

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BODY SURFACE AREA FOR CARDIO PULMONARY

This program calculates body surface area by either the method of DuBois or the method of Boyd. In both cases, the required inputs are height and weight, which may be input either in metric (cm, kg) or English (in, lb) units. Quantities in English units should be input as negative numbers. If cardiac output is given, the cardiac index can also be calculated.

Equations: Let Ht be height, Wt be weight, and BSA be the body surface area in m^2.

$$Ht\ (cm) = 2.54\ Ht\ (in.)$$

$$Wt\ (kg) = 0.45359237\ Wt\ (lb.)$$

DuBois:

$$BSA\ (m^2) = Ht\ (cm)^{0.725} \cdot Wt\ (kg)^{0.425} \cdot 7.184 \cdot 10^{-3}$$

Boyd:

$$BSA\ (m^2) = Wt\ (g)^{(0.7285\ -\ 0.0188\ \log Wt)} \cdot Ht\ (cm)^{0.3} \cdot 3.207 \cdot 10^{-4}$$

$$CI = \frac{CO}{BSA}$$

where CO is cardiac output in ℓ/min.

NOTE: The DuBois formula for BSA is undefined for children with a BSA less than 0.6 m^2. In such cases BSA should be calculated by the Boyd formula.

Reference: D. DuBois and E.F. DuBois, Clin. Cal. 10, Arch. Int. Med., 17,863,1916.

Edith Boyd, *Growth of the Surface Area of the Human Body*, U. of Minnesota Press, 1935, p. 132.

Example 1:

Patient is 176 cm in height and weights 63.5 kg. What is the body surface area by both the Du Bois and Boyd methods?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] BSA [ALPHA]	HT=?
176 [R/S]	WT=?
63.5 [R/S]	B/D?
D [R/S]	DUBOIS BSA=1.78
[B]	BOYD BSA=1.76

Example 2:

A patient 60 inches in height and 100 pounds in weight has a cardiac output of 5 1/min. Calculate the body surface area and cardiac index by Boyd. What is the cardiac index using the Du Bois BSA?

Keystrokes:	Display:
[A]	HT=?
60 [CHS] [R/S]	WT=?
100 [CHS] [R/S]	B/D?
B [R/S]	BOYD BSA=1.40
[C]	CO=( )?
5 [R/S]	CI=3.58
[D]	DUBOIS BSA=1.39
[C]	CO=5.00?
[R/S]	CI=3.60

User Instructions

SIZE: 012

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program and begin execution.		[XEQ] BSA	HT=?
2	Input the patient height (cm or -inches)	cm or -in	[R/S]	WT=?
3	Input the patient weight (kg or -lbs.)	kg or -lbs	[R/S]	B/D?
4a	To calculate body surface area by Boyd			
	method, input B.	B	[R/S]	BOYD BSA=
	or			
4b	To calculate body surface area by Du Bois			
	method, input D. Go to step 5 or 6.	D	[R/S]	DUBOIS BSA=
5	To calculate body surface area by the			
	alternate method:			
	• by Boyd method		[B]	BOYD BSA=
	• by Du Bois method		[D]	DUBOIS BSA=
6	Optional: Calculate cardiac index.			
	Recall cardiac, if previously stored.		[C]	CO=( )?
	(If cardiac output is incorrect, input			
	correct cardiac output)	CO		
	Calculate cardiac index.		[R/S]	CI=
7	For a new case press [A] and go to step 2.		[A]	HT=?

Program Listings

01♦LBL "BSA"	
02 FIX 2	Initialize
03 SF 21	
04 SF 27	
05♦LBL A	
06 "HT=?"	
07 PROMPT	Input HT
08 X>0?	Metric?
09 GTO 01	Yes
10 CHS	No, convert to cm
11 2.54	
12 *	
13♦LBL 01	
14 STO 05	STO HT
15 "WT=?"	Input WT
16 PROMPT	
17 X>0?	Metric
18 GTO 02	Yes
19 CHS	No, convert to kg
20 2.205	
21 /	
22♦LBL 02	
23 STO 06	STO WT
24 "B/D?"	Choose Boyd or
25 AON	Du Bois
26 PROMPT	
27 "⊢A"	
28 AOFF	
29 ASTO X	
30 GTO IND X	Go to calc. routine
31♦LBL D	
32♦LBL "DA"	Du Bois calc.
33 RCL 05	
34 .725	
35 Y↑X	
36 RCL 06	
37 .425	
38 Y↑X	
39 *	
40 139.2	
41 /	
42 STO 07	STO Du Bois BSA
43 "DUBOIS BSA"	
44 GTO 10	Display Du Bois BSA
45♦LBL B	
46♦LBL "BA"	Boyd calc.
47 RCL 05	
48 .3	
49 Y↑X	
50 RCL 06	
51 1 E3	
52 *	
53 ENTER↑	
54 LOG	
55 .0188	
56 *	
57 .7285	
58 -	
59 Y↑X	
60 /	
61 3118	
62 /	
63 STO 07	STO Boyd BSA
64 "BOYD BSA"	
65 GTO 10	Display Boyd BSA
66♦LBL C	
67 RCL 08	Recall CO
68 "CO"	
69 XEQ 09	
70 STO 08	STO new CO
71 RCL 07	
72 /	Calc. CI
73 "CI"	
74♦LBL 10	
75 "⊢="	Display routine
76 ARCL X	
77 AVIEW	
78 RTN	
79♦LBL 09	Recall stored CO
80 "⊢="	
81 ARCL X	
82 "⊢?"	
83 PROMPT	
84 END	

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
05	HT (cm)	55	
	WT (kg)		
	BSA (m^2)		
	CO (ℓ/min)		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 40 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21	S	Print & Display	Pause only
27	S	User mode	Reg. mode

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CARDIAC OUTPUTS

Dye Curve Cardiac Output:

This portion of the program calculates cardiac output from measurements taken directly from an indicator dilution curve. It computes the area of the first part of the curve by trapezoidal rule integration. The part after the last point is calculated from an exponential projection based on the first measured point below 65% of the peak measured point; and the first measured point after that which is below 45% of the peak. This not only avoids problems of indicator recirculation in most cases, but also limits the amount of data to be input. Thus it is important to have a measured point which is below 45% of the peak, but before recirculation becomes obvious. If this isn't possible, an approximation can be obtained by guessing at the curve without recirculation and entering values.

Equation Used:

$$CO(\ell/\text{min}) = \frac{\text{DOSE (mg)} \cdot 60\ (\text{sec/min})}{\text{CAL}(\text{mg}/(\ell \cdot \text{div})) \cdot \text{AREA (div} \cdot \text{sec)}}$$

Detailed Instructions for Dye Curve Output:

To obtain accurate results, it is important to measure the curve at frequent intervals. Generally, about ten points on the curve, equally spaced in time between onset and the 40%-of-peak point on the downslope, will be adequate. Choose and input a measurement time interval accordingly

Input the values measured from the curve (DC) and press [R/S] after each. The units of measurement are arbitrary; for example, divisions on the paper or volts, so long as the same units are used in inputting the calibration. The values are measured relative to the baseline, or starting level, of the curve. After each input entry, the display will indicate the number of points input.

As points on the downslope are input, the program compares each with the peak value. When the first point whose value is less than 65% of the peak value is found, it is stored for later use in the exponential projection as indicated by a minus sign preceding the displayed value representing the number of points input.

When a point having a value less than 45% of the peak value is input, the program automatically makes the exponential projection and displays the area under the curve, rather than the number of points entered.

At this time, input the CAL value. If indocyanine green dye is being used, it will generally be measured as milligrams of dye per liter of the patient's blood per division or unit of curve measurement. For other indicators, equivalent calibration factors must be determined.

Finally, input the dose of indicator given (for dye, this will usually be in mg.). Cardiac output in liters/min. is calculated and stored in memory.

Fick Cardiac Output:

This portion of the program computes cardiac output, stroke volume, and cardiac index by the Fick method.

Equations Used: $$CO(\ell/min) = \frac{VO_2\ (ml/min\ STPD) \cdot 100(\%)}{(C_aO_2 - C_vO_2)(vol.\%) \cdot 1000(m\ell/\ell)}$$

$$SV(ml/beat) = \frac{CO(\ell/min) \cdot 1000\ (ml/\ell)}{HR\ (beats/min)}$$

$$CI(\ell/min\ m^2) = \frac{CO(\ell/min)}{BSA(m^2)}$$

$$SI(m\ell/m^2) = \frac{SV(ml)}{BSA(m^2)}$$

Detailed Instructions for Fick Output:

If the Virtual PO_2 and O_2 Saturation and Content has just been run either or both CaO_2 and C_vO_2 will be stored. The program will automatically recall these stored values for input. Proceed as usual by inputting values or accepting recalled values for each parameter. Be sure VO_2 is in ml/min STPD.

To calculate cardiac index-assuming BSA has been previously stored, press [R/S] to recall BSA, or input the correct value. To calculate stroke volume input the heart rate. After calculating stroke volume pressing [R/S] will yield the stroke index. Pressing [R/S] again returns to the display of SV.

Example 1: (For dye curve CO)

Eight consecutive values are taken at one second intervals from an indicator dilution curve. They are as follows: 5, 20, 45, 60, 50, 38, 28, 20. The calibration is 0.2 mg/l/div. The dose is 3 mg. Calculate the cardiac output from the dye curve data.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] DYE [ALPHA]	TIME=?
1 [R/S]	DC=?

5 [R/S]	(1.00)
	DC=?
20 [R/S]	(2.00)
	DC=?
45 [R/S]	(3.00)
	DC=?
60 [R/S]	(4.00)
	DC=?
50 [R/S]	(5.00)
	DC=?
38 [R/S]	(-6.00) past 65% point.
	DC=?
28 [R/S]	(-7.00)
	DC=?
20 [R/S]	AREA=318.32
[R/S]	CAL=?
.2 [R/S]	DOSE=?
3 [R/S]	CO=2.83

Example 2: (For Fick CO)

Calculate Fick cardiac output and index, and stroke volume and index from the following data:

CaO_2 = 18 vol.%
C_VO_2 = 15 vol.%
VO_2 = 250 ml/min.STPD
BSA = $2m^2$
Heart rate = 60 BPM

Keystrokes:	Display:
[XEQ] [ALPHA] FICK [ALPHA]	CaO2=( )?
18 [R/S]	CVO2=( )?
15 [R/S]	VO2=( )?
250 [R/S]	CO=8.33
[R/S]	BSA=( )?
2 [R/S]	CI=4.17
[R/S]	HEART BPM=?
60 [R/S]	SV=138.83
[R/S]	SI=69.42

User Instructions

SIZE: 012

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program. For Dye Curve Cardiac			
	Output go to step 2. For Fick Cardiac			
	Output go to step 8.			
	DYE CURVE CARDIAC OUTPUT			
2	Begin Dye Curve Cardiac Output calculation.		[XEQ] DYE	TIME=?
3	Input the time interval.	Time (sec)	[R/S]	DC=?
4	Input the first value from the dye curve.	DC	[R/S]	(n or -n)
				DC=?
5	Repeat step 4 for all DC values. As soon			
	as DC ⩽ 45% of peak, area is calculated.			AREA=
6	Input calibration.		[R/S]*	CAL=?
		CAL (mg/(1 Div))	[R/S]	DOSE=?
7	Input dose of indicator. Cardiac Output			
	is calculated and displayed.	Dose	[R/S]	CO=
	FICK CARDIAC OUTPUT			
8	Begin Fick Cardiac Output calculation.		[XEQ] FICK	CaO2=( )?
9	Stored CaO_2 is displayed. Input correct			
	CaO_2 or, if correct, press [R/S].	CaO_2	[R/S]	CVO2=( )?
10	Stored C_VO_2 is displayed. Input correct			
	C_VO_2 or, ir correct, press [R/S].	C_VO_2	[R/S]	VO2=( )?
11	Stored VO_2 is displayed. Input correct			
	VO_2 or, if correct, press [R/S]. Fick			
	Cardiac Output is calculated and displayed.	VO_2	[R/S]	CO=
12	Recall stored BSA. Input correct BSA or,		[R/S]*	BSA=( )?
	if correct, press [R/S]. Cardiac index	BSA	[R/S]	CI=
	is calculated and displayed.		[R/S]*	HEART BPM=?
	*This [R/S] not necessary if calculator is used with printer.			

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
13	Input heart rate. Stroke volume is			
	calculated and displayed.	Rate (BPM)	[R/S]	SV=
14	Display stroke index.		[R/S]*	SI=
15	(Optional) To again display SV and SI		[R/S]	SV=
	press [R/S].		[R/S]*	SI=
16	For another Dye Curve calculation press			
	[A] and go to step 3.		[A]	TIME=?
17	For another Fick Cardiac Output press [F]			
	and go to step 9.		[F]	CaO2=( )?
	*This [R/S] not necessary if calculator is used with printer.			

Program Listings

Program	Comment
01♦LBL "DYE"	Dye Curve CO
02♦LBL A	
03 FIX 2	Initialization
04 CF 01	
05 SF 21	
06 SF 27	
07 CLX	
08 STO 00	
09 STO 01	
10 STO 02	
11 "TIME=?"	Input time int.
12 PROMPT	
13 STO 10	
14♦LBL 00	Input dye curve
15 "DC=?"	values
16 PROMPT	Count entries
17 DSE 00	
18♦LBL 11	
19 ST+ 01	Integrate
20 RCL 02	New Peak?
21 X>Y?	No
22 GTO 01	Yes
23 X<>Y	
24 STO 02	
25 X<>Y	Clear 65% flag
26 CF 01	
27♦LBL 01	
28 FS? 01	
29 GTO 02	
30 .65	
31 *	If past 65% GTO
32 X>Y?	03 else display
33 GTO 03	count
34 RCL 00	
35 CHS	
36 PSE	
37 GTO 00	Do 65% test
38♦LBL 03	
39 X<>Y	
40 STO 03	
41 RCL 00	
42 STO 04	
43 SF 01	
44 PSE	
45 GTO 00	
46♦LBL 08	Display negative
47 RCL 00	count
48 PSE	
49 GTO 00	
50♦LBL 02	

Program	Comment
51 .45	Do 45% test
52 *	
53 X<=Y?	If not past 45%
54 GTO 08	Display negative
55 RDN	count else
56 STO 02	calculate
57 RCL 04	exponential area
58 RCL 00	
59 -	
60 RCL 03	
61 RCL 02	
62 /	
63 LN	
64 /	
65 .5	
66 -	
67 *	
68 RCL 01	
69 +	
70 RCL 10	
71 *	
72 CF 01	
73 STO 02	
74 "AREA"	Display area
75 XEQ 10	
76 "CAL=?"	Input calibration
77 PROMPT	
78 ST* 02	
79 "DOSE=?"	Input dose
80 PROMPT	
81 RCL 02	Calculate CO
82 /	
83 60	
84 *	
85 RND	
86♦LBL 05	
87 STO 08	Store and display
88 "CO"	CO
89♦LBL 10	
90 "⊢="	Display routine
91 ARCL X	
92 AVIEW	
93 RTN	
94♦LBL "FICK"	Fick CO
95♦LBL F	
96 FIX 2	Initialize
97 SF 21	
98 SF 27	
99 RCL 04	
100 "CaO2"	RCL CaO_2

Program Listings

101 XEQ 09	
102 STO 04	STO CaO_2
103 RCL 03	
104 "CVO2"	RCL CVO_2
105 XEQ 09	
106 STO 03	STO CVO_2
107 RCL 02	
108 "VO2"	RCL VO_2
109 XEQ 09	
110 STO 02	STO VO_2
111 RCL 04	
112 RCL 03	
113 -	Call Fick CO
114 10	
115 *	
116 /	
117 RND	
118 XEQ 05	Display CO
119 RCL 07	
120 "BSA"	
121 XEQ 09	RCL BSA
122 STO 07	
123 RCL 08	STO BSA
124 X<>Y	
125 /	
126 "CI"	Calc. and display
127 XEQ 10	CI
128 RCL 08	
129 "HEART BPM=?"	Input Heart Rate
130 PROMPT	
131 /	
132 1 E3	Calculate
133 *	
134 STO 01	&
135♦LBL 04	
136 SF 21	Display SV
137 "SV"	
138 XEQ 10	
139 RCL 07	Calculate and
140 /	display SI
141 "SI"	
142 XEQ 10	
143 FS? 55	Repeat SV and
144 STOP	SI display
145 RCL 01	
146 GTO 04	
147♦LBL 09	RCL and store
148 "⊢="	routine
149 ARCL X	
150 "⊢?"	
151 PROMPT	
152 END	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Counter	50	
	SD/SV		
	Used/VO_2		
	D65/C_vO_2		
	-N65/CaO_2		
05		55	
	BSA		
	CO		
10	ΔT	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 53 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01	C	Past 65% point	Not past 65%
21	S	Print & Display	Pause only
27	S	User mode	Reg. mode

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

VALVE AREA

This program calculates the areas of heart valves from measured pressure gradients.

Equations Used:

$$\text{Valve Area (cm}^2\text{)} = \frac{\text{Mean Flow}}{0.0445\sqrt{\text{mean gradient}}}$$

where

$$\text{Mean Flow } (\ell/\text{sec}) = \frac{\text{CO}(\ell/\text{min.}) \cdot \text{R-R(sec)}}{\text{Valve Open Time (sec/beat)} \cdot 60 \text{ (sec/min.)}}$$

$$\text{Mitral Valve Area only} = \frac{\text{Valve Area}}{0.7}$$

Detailed User Instructions:

Choose whether the calculation is for mitral or aortic valve, then input the time duration, in seconds, of blood flow through the valve of interest; that is, the systolic ejection period (SEP) for outflow tract valves or the diastolic filling period (DFP) for A-V valves. Press [R/S].

This program permits averaging of a number of pressure gradients across the valve measured at different times while the valve is open. If the pressure gradient is to be measured at a number of different times, the time intervals should be equally spaced across the duration of the valve opening to obtain a true average. Simply input each value of pressure difference, (ΔP), in mmHg, and press [R/S] after each. The display will then show the number of input entries made. When all input entries have been made, press [R/S] without data entry. The average of all the ΔP values will be displayed ($\overline{\Delta P}$). If only one pressure gradient measurement is to be input, because averaging has been accomplished by some other means, simply input the value, press [R/S] and then press [R/S] without data entry. The input value will be displayed.

Input the R-R interval, in seconds, and press [R/S]. Cardiac output, if previously stored, will be recalled. If not, input it. Pressing [R/S] will display the valve area, in cm^2.

References: Gorlin, F., Gorlin, S.G., *Hydraulic Formula for Calculation of the Area of the Stenotic Mitral Valve, Other Cardiac Valves, and Central Circulatory Shunts*, American Heart Journal, Jan. 1957 VOL. 41, No. 1.

Hewlett-Packard Users' Library program #00207A.

Example:

DFP (mitral valve) = 0.55 sec.

ΔP = 10, 12, 8, 6, 2 mm Hg.

R-R = 0.94 sec.

CO = 5.73 1/min.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] VALVE [ALPHA]	MITRAL?
Y [R/S]	TIME=?
.55 [R/S]	PRESS DIFF=?
10 [R/S]	(1.00)
	PRESS DIFF=?
12 [R/S]	(2.00)
	PRESS DIFF=?
etc.	etc.
2 [R/S]	(5.00)
	PRESS DIFF=?
[R/S]	AVE PRESS DIFF=7.60
[R/S]	R-R=?
.94 [R/S]	CO=( )?
5.73 [R/S]	MITRAL VALVE AREA=1.90

User Instructions

SIZE: 012

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program and execute Valve Area.		[XEQ] VALVE	MITRAL?
2	If for mitral valve input	Y	[R/S]	TIME=?
2a	or, if for other valve input	N	[R/S]	TIME=?
3	Input the ejection period	seconds	[R/S]	PRESS DIFF=?
4	Input the pressure difference	ΔP, (mmHg)	[R/S]	(1.00)
				PRESS DIFF=?
4a	Repeat step 4 for all values of pressure	ΔP_n (mmHg)	[R/S]	(n)
	difference			PRESS DIFF=?
5	When all values of pressure difference			
	have been input calculate average by			
	pressing [R/S] <u>without prior data entry</u>		[R/S]	AVE PRESS DIFF=
6	Input the R-R interval		[R/S]*	R-R=?
		R-R (sec)	[R/S]	CO=( )?
7	Cardiac output, if stored, is recalled.			
	If incorrect, input correct value and press			
	[R/S]. Valve area is calculated.	CO(1/min)	[R/S]	VALVE AREA=
				or MITRAL VALVE
				AREA=
8	For a new case press [A] and go to step 2			
	*This [R/S] not required if printer is used.			

Program Listings

01♦LBL "VALVE"	Valve Area
02♦LBL A	
03 CF 00	
04 CF 01	Initialize
05 SF 21	
06 SF 27	
07 FIX 2	
08 "Y"	
09 ASTO Y	
10 "MITRAL?"	Mitral valve?
11 AON	
12 PROMPT	
13 AOFF	
14 ASTO X	
15 X=Y?	
16 SF 00	Yes
17 "TIME=?"	
18 PROMPT	Input SEP or DFP
19 STO 09	
20 CF 22	
21♦LBL 00	
22 "PRESS DIFF=?"	Input ΔP
23 PROMPT	
24 FC?C 22	
25 GTO 02	
26 FS? 01	
27 GTO 01	
28 STO 01	STO ΔP_1
29 SF 01	
30 1	STO M=1
31 STO 00	
32 PSE	
33 GTO 00	
34♦LBL 01	$\Delta P_2 - \Delta P_n$
35 ST+ 01	
36 ISG 00	
37♦LBL 15	
38 RCL 00	Display n
39 PSE	
40 GTO 00	
41♦LBL 02	
42 RCL 00	
43 ST/ 01	$\overline{\Delta P}$
44 RCL 01	
45 "AVE PRESS DIFF"	
46 XEQ 10	Display $\overline{\Delta P}$
47 RCL 09	
48 "R-R=?"	Input R-R
49 PROMPT	
50 /	
51 60	
52 *	
53 STO 10	
54 RCL 08	Recall CO
55 "CO="	
56 ARCL X	
57 "⊢?"	
58 PROMPT	
59 STO 08	STO CO
60 RCL 10	
61 /	Calc. area
62 .0445	
63 /	
64 RCL 01	
65 SQRT	
66 /	
67 CLA	
68 FC? 00	Mitral?
69 GTO 03	No
70 .7	
71 /	Mitral factor
72 "MITRAL"	
73♦LBL 03	
74 "⊢VALVE AREA"	
75♦LBL 10	
76 "⊢="	
77 ARCL X	Display routine
78 AVIEW	
79 END	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	n (counter)	50	
	$\Sigma\Delta P$, $\overline{\Delta P}$		
05		55	
	CO		
	SEP or DFP		
10	S/M, A	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 39 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	C	Mitral Calc	Other valves
01	C	Other ΔP	First ΔP
21	S	Print & Display	Pause only
27	S	User mode	Reg. mode

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CARDIAC SHUNTS

This program calculates anatomic shunts or a physiologic shunt from measured oxygen concentrations.

Anatomic Shunts:

This routine calculates left-to-right and right-to-left shunts and displays them as a percentage. The program uses the method of allegations and can calculate bi-directional shunts.

Equations Used:

$$\text{R-L shunt (\%)} = \frac{\text{(L-PUL)} - \text{(L-SYST)}}{\text{(L-PUL)} - \text{(R-SYST)}} \cdot 100$$

$$\text{L-R shunt (\%)} = \frac{\text{(R-PUL)} - \text{(R-SYST)}}{\text{(L-PUL)} - \text{(R-SYST)}} \cdot 100$$

The program assumes oxygen concentration values taken from four sites in the cardiovascular system. Since these sites may be various chambers in the heart or great vessels, they are labeled right systemic, right pulmonary, left pulmonary and left systemic. For example, suppose oxygen concentration values are known for the right atrium, pulmonary artery, left ventricle, and aorta; then the right systemic site would be the right atrium, the right pulmonary site would be the pulmonary artery, the left pulmonary site would be the left ventricle, and the left systemic site would be the aorta.

Note that it is possible to enter either oxygen contents or saturations, assuming hematocrit does not change during the sampling interval.

Physiologic Shunt and Fick Cardiac Output:

The Fick cardiac output and physiologic shunt fraction are calculated from arterial, venous and alveolar oxygen concentration and oxygen intake.

Equations Used:

$$\text{Phys. Shunt} = \frac{C_AO_2 - C_aO_2}{C_AO_2 - C_VO_2}$$

$$\text{CO } (\ell/\text{min}) = \frac{VO_2 \text{ (ml/min STPD)} \cdot 100(\%)}{(C_aO_2 - C_VO_2 \text{ (vol.\%)} \cdot 1000 \text{ (ml/}\ell\text{)}}$$

These are the standard physiologic shunt and Fick cardiac output equations. If measured O_2 saturations are used, these equations will be accurate.

If the content values have been derived from saturation estimates on PO_2 measurements for arterial and venous blood, the results should be viewed with caution unless the patient's oxygen dissociation curve has been established to be normal.

After cardiac output is calculated, stroke volume may be calculated by heart rate and multiplying by 1000 (to convert from l to ml). Alternatively, cardiac index may be calculated by dividing by body surface area.

If the program is to be used to calculate output only, it is not necessary to input C_AO_2.

References: Zimmerman, H.A., *Intravascular Catheterization*, Charles C. Thomas, Springfield, IL, 1966.

Comroe, Julius H., Jr., et al. *The Lung*, 2nd ed., Year Book Medical Publishers, Inc., Chicago, 1962, p. 345.

Hang, Sing San, et al, *From Cardiac Catheterization Data to Hemodynamic Parameters*, F.A. Davis Co., Phil., 1972, p. 21.

Example 1:

Calculate the left-to-right or right-to-left shunts for a patient having the following oxygen saturation values at the listed sites. Right atrium, 85%; pulmonary artery, 88%; left ventricle, 95%; left atrium, 93%.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] ANATOM [ALPHA]	R-SYST=?
85 [R/S]	R-PUL=?
88 [R/S]	L-PUL=?
95 [R/S]	L-SYST=?
93 [R/S]	L-R SHUNT=30.00
[R/S]	R-L SHUNT=20.00

Example 2:

Calculate physiologic shunt and Fick cardiac output from the following data:

C_AO_2 = 20 vol.%

C_aO_2 = 18 vol.%

C_VO_2 = 15 vol.%

VO_2 = 250 ml/min. STPD

Keystrokes:	Display:
[XEQ] [ALPHA] PHYS [ALPHA]	CAO2=( )?
20 [R/S]	CaO2=( )?
18 [R/S]	CVO2=( )?
15 [R/S]	PHYS SHUNT=40.00
[F]	VO2=( )?
250 [R/S]	FICK CO=8.33

User Instructions

				SIZE: 012
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program. To calculate anatomic			
	shunts go to step 2. To calculate			
	physiologic shunts go to step 3.			
	Anatomic Shunts			
2	Begin the anatomic shunt calculation		[XEQ] ANATOM	R-SYST=?
2a	Input the following oxygen concentrations:			
	Right-Systemic	R-syst	[R/S]	R-PUL=?
	Right-Pulmonary	R-pul	[R/S]	L-PUL=?
	Left-Pulmonary	L-pul	[R/S]	L-SYST=?
	Left-Systemic	L-syst	[R/S]	L-R SHUNT=
	and calculate the Left-to-Right Shunt			
2b	To calculate the Right-to-Left Shunt		[R/S]	R-L SHUNT=
2c	If no shunt is calculated display so		or,	L-R: NO SHUNT
	indicates			R-L: NO SHUNT
	Physiologic Shunts			
3	Begin the physiologic shunt calculation and			
	see stored C_AO_2		[XEQ] PHYS	CAO2=?
3a	Input correct C_AO_2, or if stored C_AO_2 is			
	correct press [R/S] without prior data	C_AO_2 or		
	entry	no input	[R/S]	CaO2=?
3b	Repeat for C_aO_2 and C_vO_2	C_aO_2 or		
		no input	[R/S]	CVO2=?
		C_vO_2 or		
		no input	[R/S]	PHYS SHUNT=
	Physiologic shunt is calculated.			
	Fick Cardiac Output (Optional)			
4	To calculate Fick Cardiac Output (if oxygen			

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	uptake is known) press [F] recalling			
	stored oxygen uptake, input correct			
	oxygen uptake or if correct press [R/S]			
	without prior data entry. Fick cardiac			
	output is then calculated.		[F]	VO2=( )?
		VO_2 or no input	[R/S]	FICK CO=
5	For a new anatomical shunt press [A] and			
	go to step 2		[A]	R-SYST=?
6	For a new physiologic shunt press [C] and			
	go to step 3		[C]	CAO2=( )?
6a	For a new Fick cardiac output press [F]			
	and go to step 4		[F]	VO2=( )?

Program Listings

Listing	Comments
01♦LBL "ANATOM"	Anatomic shunts
02♦LBL A	
03 FIX 2	Initialize
04 SF 21	
05 SF 27	
06 SF 01	
07 "R-SYST=?"	
08 PROMPT	
09 STO 00	
10 "R-PUL=?"	Input right and
11 PROMPT	left pulmonary
12 STO 02	and systemic
13 "L-PUL=?"	O_2 concentra-
14 PROMPT	tions
15 STO 05	
16 "L-SYST=?"	
17 PROMPT	
18 STO 06	
19 SF 00	
20♦LBL 00	Toggle for L-R
21 FS?C 00	to R-L calc.
22 GTO 02	
23 SF 00	
24 RCL 05	Calc. R-L shunt
25 RCL 06	
26 -	
27 RCL 05	
28 RCL 00	
29 -	
30 /	
31 "R-L"	
32 GTO 03	
33♦LBL 02	Calc. L-R shunt
34 RCL 02	
35 RCL 00	
36 -	
37 RCL 05	
38 RCL 00	
39 -	
40 /	
41 "L-R"	
42♦LBL 03	
43 1 E2	
44 *	
45 X<=0?	Is result pos.
46 GTO 01	No
47 XEQ 10	
48♦LBL 04	Yes, display
49 FS? 55	shunt
50 STOP	Toggle for
51 GTO 00	
52♦LBL 01	
53 "├: NO SHUNT"	Display: no shunt
54 AVIEW	
55 FS? 01	
56 GTO 04	Go to toggle
57 RTN	
58♦LBL 10	
59 "├ SHUNT="	Display shunt
60 ARCL X	
61 AVIEW	
62 FS? 01	
63 GTO 04	Go to toggle
64 RTN	
65♦LBL "PHYS"	Calc. physiologic shunt
66♦LBL C	
67 FIX 2	
68 SF 21	Initialize
69 SF 27	
70 CF 01	
71 RCL 01	
72 "CAO2"	RCL C_AO_2
73 XEQ 09	
74 STO 01	STO C_AO_2
75 RCL 04	
76 "CaO2"	RCL C_aO_2
77 XEQ 09	
78 STO 04	STO C_aO_2
79 RCL 03	
80 "CVO2"	RCL C_VO_2
81 XEQ 09	
82 STO 03	STO C_VO_2
83 RCL 01	Calculate shunt
84 RCL 04	
85 -	
86 RCL 01	
87 RCL 03	
88 -	
89 /	
90 1 E2	
91 *	
92 "PHYS"	
93 X<=0?	
94 GTO 01	No shunt

Program Listings

95 GTO 10	Display shunt
96♦LBL F	Calc. Fick CO
97 RCL 02	
98 "VO2"	
99 XEQ 09	RCL VO_2
100 STO 02	STO VO_2
101 RCL 04	
102 RCL 03	Calc. CO
103 -	
104 /	
105 10	
106 /	
107 STO 08	
108 "FICK CO ="	
109 ARCL X	
110 AVIEW	Output Fick CO
111 RTN	
112♦LBL 09	
113 "⊢="	Display of recalled data
114 ARCL X	
115 "⊢?"	
116 PROMPT	
117 END	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	R-Syst	50	
	C_aO_2		
	R-Pulm/VO_2		
	$C_{\bar{v}}O_2$		
	C_aO_2		
05	L-Pulm	55	
	L-Syst		
	BSA		
	CO		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 50 USER MODE

ENG ___ FIX 2 SCI ___ ON ___ OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	S	L-R shunt calc.	R-L shunt calc.
01	S	Anatomic	Physiologic
21	S	Print & Display	Pause only
27	S	User mode	Reg. mode

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CONTRACTILITY AND STROKE WORK

Contractility:

This portion of the program, entitled "Vmax," calculates the indices of left ventricular contractility based on pressure rise during isovolumetric contraction.

Equations Used:

P_N = most recently entered pressure (mmHg)

P_{N-1} = next previously entered pressure

Δt = time interval between pressure measurements (sec)

P_P = pressure at which dP/dt/P is calculated

$$\Delta P = P_N - P_{N-1}$$

$$\frac{dP}{dt} = \frac{\Delta P}{\Delta t} \text{ mmHg/sec}$$

$$P_P = \frac{P_N + P_{N-1}}{2}$$

$$dP/dt/P = \frac{dP/dt}{P_P} \text{ sec}^{-1}$$

P_M = P_P where dP/dt/P is a maximum

$$V_{MAX} = \frac{1}{30} \; \frac{(P_{P\ LAST} \cdot \text{MAX } dP/dt/P) - (P_M \cdot dP/dt/P_{\ LAST})}{P_{P\ LAST} - P_M}$$

dP/dt is calculated as the difference between successive pressure inputs divided by the time interval Δt. The largest value found is stored as maximum dP/dt.

dP/dt/P is calculated for each pair of successive inputs, by first determining dP/dt as above, then dividing by the mean of the two pressures. The largest value found is stored as maximum dP/dt/P.

V_{MAX} is found in this program by a linear projection of the downslope of the dP/dt/P vs. P curve back to P = 0, and by dividing the resulting dP/dt/P by 30. The projection is based on the point at which the maximum dP/dt/P was found, and the last point input. The constant is controversial, values between about 28 and 32 having appeared in the literature. The value 30 is used in this program.

Detailed Instructions:

The indices of left ventricular contractility calculated by this program are based on the pressure rise during isovolumetric contraction. Measurements, equally spaced in time, should be input for the isovolumetric phase only. Inputting values from the systolic ejection period can cause significant errors. Generally, between 5 and 10 pressure measurements should be input, and the time interval between measurements, Δt, chosen accordingly. Too few measurements will cause the maximum values to be missed. Too many will introduce excessive "noise" resulting in errors.

After each input except the first, dP/dt/P for the two most recent points will be displayed with a pause. When all inputs have been made the results: maximum dP/dt, maximum dP/dt/P and V_{MAX}, maximum velocity of the contractile element at zero pressure in circumferences or lengths/sec., are displayed.

If the contractility parameters are to be calculated using developed pressure, or any pressure reference other than zero, perform the subtraction before entering pressure values.

Stroke Work:

This routine calculates stroke work (SW) and stroke work index (SWI). For stroke work based on systolic minus end-diastolic pressure, perform subtraction before data input.

Equations Used:

$$SW(gm \cdot m) = \frac{13.6 \cdot P(mmHg) \cdot CO(\ell/min) \cdot R\text{-}R(sec)}{60\ (sec/min)}$$

$$SWI(gm/m) = \frac{SW(gm \cdot m)}{BSA(m^2)}$$

Detailed Instructions:

The mean systolic pressure, $\overline{P}$, is required for stroke work calculation. The program will average pressures measured at equal time intervals through systole to obtain the mean. When all inputs have been made, press [R/S] without prior data entry to obtain the mean systolic pressure.

If averaging is accomplished by other means, only a single value is input. If an error is made in the pressure inputs, restart program by pressing [B] and rekey the input data.

Reference: Yang, Sing San, et al, *From Cardiac Catheterization Data to Hemodynamic Parameters*, F.A. Davis Co., Phil., 1972.

Example 1:

Find maximum dP/dt, maximum dP/dt/P and maximum ventricular contractility if the time interval is 0.005 seconds and P_N is 10, 20, 40, 60, and 80 mmHg.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] VMAX [ALPHA]	TIME INT.=?
.005 [R/S]	P1=?
10 [R/S]	P2=?
20 [R/S]	(133.33)
	P3=?
40 [R/S]	(133.33)
	P4=?
60 [R/S]	(80.00)
	P5=?
80 [R/S]	(57.14)
	P6=?
[R/S]	MAX dP/dT=4000
[R/S]	MAX dP/dT/P=133.3
[R/S]	VMAX=5.14

Example 2:

P_{SYS} = 100,110 mmHg

R-R = 1 sec

CO = 5 l/min.

BSA = $2m^2$

Keystrokes:	Display:
[XEQ] [ALPHA] WORK [ALPHA]	PSYST=?
100 [R/S]	PSYST=?
110 [R/S]	PSYST=?
[R/S]	AVE P=105.00
[R/S]	R-R=?
1 [R/S]	CO=( )?
5 [R/S]	STROKE WORK=119.00
[R/S]	BSA=( )?
2 [R/S]	SW INDEX=59.50

User Instructions

				SIZE: 012
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program. For Contractibility			
	(Vmax) calculation go to step 2. For			
	stroke work go to step 8.			
	CONTRACTIBILITY			
2	Begin contractibility calculation.		[XEQ] VMAX	TIME INT.=?
3	Input the time interval.	ΔT (sec)	[R/S]	P1=?
4	Input first pressure reading.	P1 (mmHg)	[R/S]	P(n)=?
5	Input next pressure reading.	P2 (mmHg)	[R/S]	(dP/dt/P)
	(dP/dt/P for two most recent points is			P(n+1)=?
	displayed with a pause).			
6	Repeat step 5 for remainder of pressure			
	readings. When all readings are input,			
	press [R/S] *without prior data entry*.	no entry	[R/S]	MAX dP/dT=
	Maximum dP/dt (mmHg/sec) is displayed.			
7	Calculate maximum dP/dt/P(sec^{-1}) and Vmax		[R/S]*	MAX dP/dT/P=
	(circ/sec).		[R/S]*	VMAX=
	STROKE WORK			
8	Begin stroke work calculation.		[XEQ] WORK	PSYST=?
9	Input systolic pressure.	P_{sys} (mmHg)	[R/S]	PSYST=?
10	Repeat step 9 for all valves of Psys.			
	After all valves have been input press			
	[R/S] *without prior data entry*.	no entry	[R/S]	AVE P=
	Average Psys is displayed.			
11	Input R-R interval.		[R/S]	R-R=?
		R-R (sec)	[R/S]	CO=( )?
	*This [R/S] not necessary if calculator is used with printer.			

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
12	Stored cardiac output is displayed input			
	correct cardiac output or, if correct,	CO		
	press [R/S]. Stroke work is calculated		[R/S]	STROKE WORK=
13	Recall stored body surfact Area. Input		[R/S]	BSA=( )?
	correct value or, if correct, press [R/S]	BSA		
	stroke work index is displayed.		[R/S]	SW INDEX=

Program Listings

Program	Comment
01◆LBL "VMAX"	
02◆LBL A	Contractility Initialization
03 SF 21	
04 SF 27	ΔT input
05 "TIME INT.=?"	
06 PROMPT	
07 STO 06	
08 CLX	
09 STO 00	
10 STO 01	
11 STO 03	
12 STO 04	
13 STO 05	
14 CF 22	
15◆LBL 00	
16 ISG 00	
17◆LBL 11	
18 CLX	
19 FIX 0	
20 CF 29	
21 "P"	
22 ARCL 00	Input P_{1--m}
23 "⊢=?"	
24 PROMPT	
25 FC?C 22	
26 GTO 01	
27 ENTER↑	
28 X<> 01	
29 X=0?	
30 GTO 00	
31 -	
32 ENTER↑	Calc dP/dt
33 ENTER↑	
34 RCL 06	
35 /	
36 RCL 03	
37 X<>Y	
38 X>Y?	
39 STO 03	Save MAX dP/dt
40 RCL 01	
41 R↑	
42 2	
43 /	
44 -	Save P_1
45 STO 09	
46 /	
47 STO 02	
48 RCL 04	
49 X<>Y	

Program	Comment
50 FIX 2	Display dP/dt/P
51 PSE	
52 X<=Y?	
53 GTO 00	
54 STO 04	
55 LASTX	
56 STO 05	
57 RDN	
58 GTO 00	
59◆LBL 01	
60 RCL 03	
61 "MAX dP/dT"	
62 ASTO 10	
63 XEQ 10	Display MAXdP/dt
64 SF 29	
65 RCL 04	
66 FIX 1	
67 CLA	
68 ARCL 10	
69 "⊢/dT/P"	
70 XEQ 10	Display MAXdP/dt/P
71 RCL 09	
72 RCL 04	
73 *	
74 RCL 05	Calc V_{max}
75 RCL 02	
76 *	
77 -	
78 RCL 09	
79 RCL 05	
80 -	
81 X≠0?	
82 /	
83 30	
84 /	
85 FIX 2	
86 "VMAX"	Display VMAX
87◆LBL 10	Display Routine
88 "⊢="	
89 ARCL X	
90 AVIEW	
91 RTN	WORK
92◆LBL "WORK"	
93◆LBL 8	
94 FIX 2	
95 CF 01	
96 CF 22	Initialization
97 SF 21	
98 SF 27	

Program Listings

Program	Comments
99 SF 29	
100 0	
101 STO 00	Input P_{syst}
102◆LBL 02	
103 "PSYST"	
104 "⊢=?"	Data input?
105 PROMPT	No
106 FC?C 22	1st input
107 GTO 04	No
108 ISG 00	
109◆LBL 11	Initial input
110 FS? 01	
111 GTO 03	
112 STO 01	
113 SF 01	
114 GTO 02	
115◆LBL 03	
116 ST+ 01	
117 GTO 02	
118◆LBL 04	
119 CF 01	
120 RCL 00	
121 ST/ 01	AVE P
122 RCL 01	
123 "AVE P"	
124 XEQ 10	Display AVE P
125 "R-R=?"	Input R-R Int.
126 PROMPT	
127 STO 10	
128 RCL 08	
129 "CO"	Recall CO
130 XEQ 09	
131 STO 08	Calc stroke work
132 RCL 10	
133 *	
134 60	
135 /	
136 RCL 01	
137 *	
138 13.6	
139 *	
140 STO 09	
141 "STROKE WORK"	
142 XEQ 10	Display stroke work
143 RCL 07	
144 "BSA"	Recall BSA
145 XEQ 09	
146 STO 07	
147 RCL 09	
148 X<>Y	
149 /	
150 "SW INDEX"	Display stroke work index
151 GTO 10	Recall stored data
152◆LBL 09	
153 "⊢="	
154 ARCL X	
155 "⊢?"	
156 PROMPT	
157 END	

70

80

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	Pn / $\Sigma P, \overline{P}$		
	dP/dt/P		
	MAXdP/dt		
	MAX dP/dt/P		
05	Pn	55	
	Δt		
	BSA		
	CO		
	Pp/SW		
10	USED/R-R	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 57 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01	C	Input P_2 to Pn	First input
21	S	Print display	Pause only
22	C	Digit entry	No digit entry
27	S	User mode	Reg. mode
29	C	Radix	No Radix

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

NOTES

NOTES

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41C programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41C, HP has an extensive library of "Application Pacs". These programs transform your HP-41C into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, minature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41C.

You can choose from:

Aviation
Clinical Lab
Circuit Analysis
Financial Decisions
Mathematics
Structural Analysis
Surveying
Securities
Statistics
Stress Analysis
Games
Home Management
Machine Design
Navigation
Real Estate
Thermal and Transport Science

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41C Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

* Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales
Home Construction Estimating
Lending, Saving and Leasing
Real Estate
Small Business
Geometry
High-Level Math
Test Statistics
Antennas
Chemical Engineering
Control Systems
Electrical Engineering
Fluid Dynamics and Hydraulics
Civil Engineering
Heating, Ventilating & Air Conditioning
Mechanical Engineering
Solar Engineering
Calendars
Cardiac/Pulmonary
Chemistry
Games
Optometry I (General)
Optometry II (Contact Lens)
Physics
Surveying

* Some books require additional memory modules to accomodate all programs.

CARDIAC/PULMONARY

HEWLETT hp PACKARD

00041-90097 Printed in U.S.A.

HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Bar Codes

Cardiac/Pulmonary

CARDIAC/PULMONARY

PULMONARY FUNCTIONS/
VITAL CAPACITY
PROGRAM REGISTERS NEEDED: 89

ROW 1 (1 - 3)

ROW 2 (3 - 9)

ROW 3 (10 - 17)

ROW 4 (17 - 22)

ROW 5 (22 - 29)

ROW 6 (30 - 34)

ROW 7 (34 - 40)

ROW 8 (40 - 46)

ROW 9 (46 - 50)

ROW 10 (51 - 54)

ROW 11 (55 - 58)

ROW 12 (59 - 64)

ROW 13 (65 - 70)

ROW 14 (70 - 74)

ROW 15 (74 - 78)

ROW 16 (78 - 82)

ROW 17 (82 - 88)

ROW 18 (88 - 95)

ROW 19 (96 - 105)

ROW 20 (106 - 111)

ROW 21 (111 - 119)

ROW 22 (119 - 124)

ROW 23 (125 - 127)

ROW 24 (128 - 134)

ROW 25 (134 - 141)

ROW 26 (141 - 147)

ROW 27 (147 - 151)

ROW 28 (151 - 156)

ROW 29 (157 - 161)

ROW 30 (162 - 167)

ROW 31 (167 - 171)

ROW 32 (171 - 175)

ROW 33 (176 - 183)

ROW 34 (183 - 189)

ROW 35 (189 - 192)

ROW 36 (192 - 199)

ROW 37 (199 - 205)

ROW 38 (205 - 209)

ROW 39 (209 - 214)

ROW 40 (214 - 221)

ROW 41 (222 - 227)

ROW 42 (228 - 233)

ROW 43 (233 - 239)

ROW 44 (240 - 242)

ROW 45 (242 - 251)

ROW 46 (251 - 254)

ROW 47 (255 - 263)

ROW 48 (264 - 267)

LUNG DIFFUSION AND ARTERIAL
CO2 NORMALIZATION
PROGRAM REGISTERS NEEDED: 33

ROW 1 (1 – 4)

ROW 2 (4 – 7)

ROW 3 (7 – 12)

ROW 4 (13 – 16)

ROW 5 (16 – 20)

ROW 6 (21 – 26)

ROW 7 (27 – 37)

ROW 8 (37 – 40)

ROW 9 (40 – 45)

ROW 10 (46 – 53)

ROW 11 (53 – 59)

ROW 12 (59 – 61)

ROW 13 (61 – 68)

ROW 14 (68 – 72)

ROW 15 (72 – 78)

ROW 16 (78 – 88)

ROW 17 (89 – 95)

ROW 18 (96 – 99)

VENTILATOR SETUP & CORRECTIONS

PROGRAM REGISTERS NEEDED: 49

HEWLETT PACKARD
SOLUTION BOOK:
CARDIAC/PULMONARY

ROW 1 (1 - 4)

ROW 2 (4 - 10)

ROW 3 (10 - 16)

ROW 4 (16 - 23)

ROW 5 (23 - 29)

ROW 6 (30 - 37)

ROW 7 (38 - 44)

ROW 8 (45 - 51)

ROW 9 (52 - 62)

ROW 10 (62 - 67)

ROW 11 (67 - 72)

ROW 12 (72 - 76)

ROW 13 (77 - 85)

ROW 14 (85 - 93)

ROW 15 (94 - 98)

ROW 16 (98 - 103)

ROW 17 (104 - 112)

ROW 18 (112 - 116)

ROW 19 (116 - 121)

ROW 20 (122 - 125)

ROW 21 (125 - 129)

ROW 22 (129 - 133)

ROW 23 (133 - 135)

ROW 24 (135 - 142)

ROW 25 (142 - 144)

ROW 26 (144 - 149)

ROW 27 (149 - 149)

BLOOD CHEMISTRY I

PROGRAM REGISTERS NEEDED: 50

HEWLETT PACKARD
SOLUTION BOOK:
CARDIAC/PULMONARY

ROW 1 (1 - 4)

ROW 2 (4 - 9)

ROW 3 (10 - 15)

ROW 4 (16 - 22)

ROW 5 (23 - 28)

ROW 6 (29 - 39)

ROW 7 (40 - 46)

ROW 8 (46 - 48)

ROW 9 (49 - 54)

ROW 10 (54 - 60)

ROW 11 (60 - 67)

ROW 12 (68 - 75)

ROW 13 (76 - 83)

ROW 14 (83 - 92)

ROW 15 (92 - 100)

ROW 16 (101 - 106)

ROW 17 (107 - 113)

ROW 18 (114 - 119)

ROW 19 (119 - 123)

ROW 20 (124 - 131)

ROW 21 (132 - 135)

ROW 22 (136 - 143)

ROW 23 (144 - 150)

ROW 24 (151 - 158)

ROW 25 (159 - 166)

ROW 26 (166 - 173)

ROW 27 (173 - 175)

BLOOD CHEMISTRY II

PROGRAM REGISTERS NEEDED: 27

HEWLETT PACKARD
SOLUTION BOOK:
CARDIAC/PULMONARY

ROW 1 (1 - 4)

ROW 2 (4 - 9)

ROW 3 (10 - 15)

ROW 4 (16 - 23)

ROW 5 (23 - 28)

ROW 6 (29 - 34)

ROW 7 (34 - 39)

ROW 8 (39 - 44)

ROW 9 (44 - 50)

ROW 10 (51 - 57)

ROW 11 (57 - 65)

ROW 12 (66 - 73)

ROW 13 (73 - 79)

ROW 14 (79 - 85)

ROW 15 (86 - 88)

BODY SURFACE AREA FOR
CARDIO PULMONARY
PROGRAM REGISTERS NEEDED: 28

HEWLETT PACKARD
SOLUTION BOOK:
CARDIAC/PULMONARY

ROW 1 (1 – 4)

ROW 2 (5 – 11)

ROW 3 (11 – 17)

ROW 4 (18 – 24)

ROW 5 (24 – 30)

ROW 6 (31 – 34)

ROW 7 (35 – 40)

ROW 8 (41 – 43)

ROW 9 (44 – 48)

ROW 10 (49 – 55)

ROW 11 (56 – 61)

ROW 12 (62 – 65)

ROW 13 (66 – 73)

ROW 14 (73 – 80)

ROW 15 (80 – 84)

CARDIAC OUTPUTS

PROGRAM REGISTERS NEEDED: 41

HEWLETT PACKARD
SOLUTION BOOK:
CARDIAC/PULMONARY

ROW 1 (1 - 4)

ROW 2 (5 - 11)

ROW 3 (11 - 17)

ROW 4 (18 - 27)

ROW 5 (28 - 35)

ROW 6 (36 - 45)

ROW 7 (46 - 54)

ROW 8 (55 - 66)

ROW 9 (67 - 74)

ROW 10 (75 - 79)

ROW 11 (79 - 85)

ROW 12 (86 - 93)

ROW 13 (94 - 97)

ROW 14 (97 - 102)

ROW 15 (103 - 108)

ROW 16 (108 - 116)

ROW 17 (117 - 122)

ROW 18 (123 - 129)

ROW 19 (129 – 132)

ROW 20 (132 – 139)

ROW 21 (140 – 146)

ROW 22 (147 – 152)

VALVE AREA

PROGRAM REGISTERS NEEDED: 28

HEWLETT PACKARD
SOLUTION BOOK:
CARDIAC/PULMONARY

ROW 1 (1 - 3)

ROW 2 (4 - 10)

ROW 3 (10 - 15)

ROW 4 (16 - 20)

ROW 5 (21 - 22)

ROW 6 (22 - 29)

ROW 7 (30 - 38)

ROW 8 (39 - 45)

ROW 9 (45 - 46)

ROW 10 (47 - 53)

ROW 11 (54 - 60)

ROW 12 (61 - 68)

ROW 13 (69 - 72)

ROW 14 (73 - 74)

ROW 15 (75 - 79)

CARDIAC SHUNTS

PROGRAM REGISTERS NEEDED: 38

HEWLETT PACKARD
SOLUTION BOOK:
CARDIAC/PULMONARY

ROW 1 (1 - 3)

ROW 2 (3 - 7)

ROW 3 (7 - 10)

ROW 4 (11 - 16)

ROW 5 (16 - 20)

ROW 6 (21 - 30)

ROW 7 (31 - 39)

ROW 8 (40 - 46)

ROW 9 (47 - 53)

ROW 10 (53 - 56)

ROW 11 (56 - 60)

ROW 12 (60 - 65)

ROW 13 (65 - 71)

ROW 14 (72 - 76)

ROW 15 (76 - 81)

ROW 16 (81 - 90)

ROW 17 (91 - 96)

ROW 18 (97 - 104)

ROW 19 (105 - 108)

ROW 20 (109 - 115)

ROW 21 (116 - 117)

ROW 1 (1 - 4)

ROW 2 (4 - 5)

ROW 3 (6 - 16)

ROW 4 (17 - 23)

ROW 5 (23 - 31)

ROW 6 (32 - 44)

ROW 7 (45 - 55)

ROW 8 (56 - 61)

ROW 9 (61 - 66)

ROW 10 (67 - 70)

ROW 11 (71 - 83)

ROW 12 (83 - 88)

ROW 13 (89 - 93)

ROW 14 (93 - 99)

ROW 15 (100 - 104)

ROW 16 (105 - 112)

ROW 17 (113 - 120)

ROW 18 (121 - 125)

ROW 19 (125 - 130)

ROW 20 (130 - 138)

ROW 21 (139 - 141)

ROW 22 (141 - 146)

ROW 23 (147 - 151)

ROW 24 (151 - 157)

ROW 25 (157 - 157)

HEWLETT
PACKARD